WARDEQ - A FORTRAN IMPLEMENTATION OF THE WARD TYPE POWER EQUIVALENTS

J.W. Bandler, M.A. El-Kady and G. Centkowski

SOS-83-24-L

October 1983

WARDEQ - A FORTRAN IMPLEMENTATION OF THE WARD TYPE POWER EQUIVALENTS

J.W. Bandler, M.A. El-Kady and G. Centkowski

Abstract

This document contains a listing of the computer package WARDEQ described in [2]. WARDEQ is a package of forty subroutines for power systems reduction using the Ward type load flow equivalents. The package has been developed for the CDC 170/730 system with the NOS 1.4 level 552 operating system and the Fortran Extended (FTN) version 4.8 compiler. The listing contains a total of 2783 lines, of which 1229 are comments.

This work was supported by the Natural Sciences and Engineering Research Council of Canada under Grants A7239, A1708 and G0647.

The authors are with the Simulation Optimization Systems Research Laboratory and the Department of Electrical and Computer Engineering, McMaster University, Hamilton, Canada L8S 4L7.

M.A. El-Kady is also with Ontario Hydro, Toronto, Canada.

G. Centkowski is on leave from the Institute of Electronics Fundamentals, Technical University of Warsaw, Warsaw, Poland.

I. INTRODUCTION

WARDEQ is a package of forty Fortran subroutines for power systems reduction using external equivalents. The package implements five Ward type equivalent methods [1] and it also performs a contingency analysis of the reduced power system. The user's manual of the WARDEQ package together with illustrative examples are found in [2].

The package WARDEQ and its documentation have been developed for the CDC 170/730 system with the NOS 1.4 level 552 operating system. At McMaster University it is available in the form of a library relocatable subroutines. The library is available as a group indirect file LIBWARD under the charge RJWBAND.

To perform a contingency analysis of the reduced power system the package calls appropriate subroutines of the TTM1 [3] and the MA28 [4] packages, and therefore these packages must be available when the contingency analysis option is performed.

II. REFERENCES

[1] J.W. Bandler, M.A. El-Kady and G. Centkowski, "Review of the Ward class of external equivalents for power systems", Department of Electrical and Computer Engineering, McMaster University, Hamilton, Canada, Report SOS-83-23-R, 1983.

[2] J.W. Bandler, M.A. El-Kady and G. Centkowski, "WARDEQ - A Fortran implementation of the Ward type power equivalents", Department of Electrical and Computer Engineering, McMaster University, Hamilton, Canada, Report SOS-83-24-U, 1983.

[3] J.W. Bandler, M.A. El-Kady and J. Wojciechowski, "TTM1 - A Fortran implementation of the Tellegen theorem method to power system simulation and design", Department of Electrical and Computer Engineering, McMaster University, Hamilton, Canada, Report SOS-82-12-U2, 1983.

[4] I.S. Duff, "MA28 - a set of Fortran subroutines for sparse unsymmetric linear equations", Computer Science and Systems Division, AERE Harwell, Oxfordshire, England, Report R.8730, 1980.

III. LISTING OF THE WARDEQ PACKAGE

Subroutine	Number of lines (source text)	Number of words (compiled code)	Description (page of [2])	Listing (page)
ASUB	132	364		5
BELONG	18	34		7
CHNGAD	31	47		7
CHNGC	22	31		7
CHNGI	14	14		8
CHNGJ	21	24		8
CHNGP	24	30		8
CHNUMB	66	104		9
EQBCV	171	445		10
EQCCC	61	240		12
EQCOMP	275	1412		13
EQCONTI	148	501	18	17
EQCURR	19	35		20
EQDDD	48	65		20
EQDEIE	107	355	44	22
EQDRC1	71	113		23
EQEXEQ	183	736	46	24
EQFORM	128	454	49	27
EQINBR	70	121		29
EQMOSO	41	47		30
EQINRC	43	111		31
EQREAD	154	774	52	31
EQREEY	21	31		35
EQREMS	30	31		35
EQWAES	48	53		35
EQWARD	146	531	8	36
EQWEES	164	433		38
GAUSEL	41	76		41
INTERV	53	61		41
MOVE	36	41		42

Subroutine	Number of lines (source text)	Number of words (compiled code)	Description (page of [2])	Listing (page)
NOTE	14	25		43
ORDEVE	65	55		43
PRINAD	24	125		44
PRINTNC	43	236		44
RDATA	32	67	53	45
SETBIN	14	27		46
SETBIT	12	13		46
SETVEC	108	241		46
SPAMAT	72	117		48
ZERO	13	12		49

```
C
C
      SUBROUTINE ASUB (N,NEXT,NEL,JRYT,ICYT,YT,V,AK,NCOL,NROW,IFLAG,IWRI
     1TE)
C
C       THIS SUBROUTINE PERFORMS GAUSSIAN ELIMINATION OF THE "NEXT"
C       EQUATIONS FOR THE SYSTEM OF "N" LINER EQUATIONS DESCRIBED BY
C       THE MATRIX WHICH IS STORED IN SPARSE FORM (VECTORS JRYT,ICYT,YT)
C       AFTER PERFORMING THE ELIMINATION THE REDUCED SYSTEM OF THE
C       EQUATIONS IS STORED IN THE COEFFICIENT MATRIX (VECTORS AK,NROW,
C       NCOL)
C
C       N           NUMBER OF EQUATIONS
C       NEXT        NUMBER OF EQUATIONS FOR WHICH THE ELIMINATION IS
C                   PERFORMED
C       NEL         NUMBER OF NONZERO ELEMENTS OF THE COEFFICIENT MATRIX
C       V           VECTOR OF LENGTH N. A RIGHT HAND SIDE VECTOR
C       IFLAG       RETURN FLAG. FOR SUCCESSFUL RETURN IFLAG=0
C
      INTEGER ICYT(1),JRYT(1),NCOL(1),NROW(1)
      COMPLEX YT(1),V(1),AK(1),Y
C
C       THE DESCRIPTION OF THE SYSTEM OF EQUATIONS IS PRINTED OUT BY
C       THE SUBROUTINE PRINTNC
C
      IOTP=6
      IF (IWRITE.EQ.4) CALL PRINTNC (NEL,NROW,NCOL,AK,IOTP)
      NN=N
      NYT=NEL
      DO 80 II=1,NEXT
C
C       IN THIS LOOP THE ELIMINATION OF THE "NEXT" ROWS FROM THE MATRIX
C       DESCRIBING THE SYSTEM OF EQUATIONS (VECTORS YT,JRYT,ICYT)
C       IS PERFORMED
C
      I=1
C
C       MODIFY THE VALUE OF THE ELEMENTS IN THE ITH ROW
C
      J1=JRYT(I)+1
      J2=JRYT(I+1)-1
      J0=J1-1
      L=0
      Y=YT(J0)
      V(I)=V(I)/Y
      IF (J1-1.EQ.J2) GO TO 70
      DO 10 K=J0,J2
C
C       STORE THE NEW VALUE OF THE ELEMENTS FROM THE ITH ROW IN THE
C       VECTOR AK
C
      YT(K)=YT(K)/Y
      L=L+1
      AK(L)=YT(K)
   10 CONTINUE
      DO 60 J3=J1,J2
C
C       IN THIS LOOP THE ROWS WITH THE ITH NONZERO ELEMENT ARE
C       DETERMINED AND MODIFIED
C
      J=ICYT(J3)
      K1=JRYT(J)+1
      K2=JRYT(J+1)-1
      DO 20 K=K1,K2
      IF (ICYT(K).EQ.I) GO TO 30
```

```
   20 CONTINUE
      IFLAG=-10
      WRITE (6,130) IFLAG
      RETURN
   30 CONTINUE
      Y=YT(K)
      L=0
C
C        MODIFY THE VALUE OF THE ELEMENTS OF THE ITH ROW
C
      DO 40 K=J0,J2
      L=L+1
      YT(K)=AK(L)*Y
   40 CONTINUE
C
C        MODIFY THE JTH ROW
C
      V(J)=V(J)/Y
      CALL SETVEC (N,NYT,JRYT,ICYT,YT,I,J,NEL,IFLAG)
      V(J)=V(J)-V(I)
      IF (IFLAG.GE.0) GO TO 50
      RETURN
   50 CONTINUE
   60 CONTINUE
   70 CONTINUE
C
C        REMOVE THE ITH ROW FROM THE MATRIX DESCRIBING THE SYSTEM OF
C        EQUATIONS (VECTORS YT,JRYT,ICYT)
C
      CALL MOVE (N,NYT,YT,JRYT,ICYT)
C
   80 CONTINUE
C
C        END OF THE ELIMINATION
C
C        FORM THE COEFFICIENT MATRIX (VECTORS AK,NROW,NCOL) FROM THE
C        VECTORS YR,JRYT,ICYT
C
      NEL=0
      DO 90 I=1,N
      J1=JRYT(I)
      NEL=NEL+1
      AK(NEL)=YT(J1)
      NROW(NEL)=I
      NCOL(NEL)=I
   90 CONTINUE
      IF (N.EQ.1) RETURN
      DO 110 I=1,N
      J1=JRYT(I)+1
      J2=JRYT(I+1)-1
      IF (J1-1.EQ.J2) GO TO 110
      DO 100 J=J1,J2
      NEL=NEL+1
      NROW(NEL)=I
      NCOL(NEL)=ICYT(J)
      AK(NEL)=YT(J)
  100 CONTINUE
  110 CONTINUE
      N=NN
      IF (IWRITE.NE.4) RETURN
      WRITE (6,120)
      CALL PRINTNC (NEL,NROW,NCOL,AK,IOTP)
  120 FORMAT (1H /" EXTERNAL DESCRIPTION OF THE SYSTEM"/)
  130 FORMAT (2X,"IFLAG FROM ASUB IFLAG= ",I3,3X,"AN ERROR IN THE MATRIX
     1 ICYT")
```

```
      RETURN
      END
C
C
      LOGICAL FUNCTIONBELONG(IEL,IB)
C
C       THIS LOGICAL FUNCTION CHECKS IF THE ELEMENT IEL HAS BEEN
C       CODED IN THE VECTOR IB
C
      INTEGER IB(1)
C
      BELONG=.FALSE.
      IX=(IEL-1)/50
      IY=IEL-IX*50
      J=IX+1
      K=0
      CALL SETBIT (K,IY)
      BELONG=OR(IB(J),K).EQ.IB(J)
      RETURN
      END
C
C
      SUBROUTINE CHNGAD (N,NEL,JRYT,ICYT,YT,NROW,NCOL,AK,NEWNU)
C
C       THIS SUBROUTINE CHANGES A BUS ADMITTANCE MATRIX STORED IN A
C       SPARSE FORM (VECTORS YT,JRYT,ICYT) TO THE FORM OF A COEFFICIENT
C       MATRIX AK (VECTORS AK,NROW,NCOL) AND PERFORMS RENUMBERING OF THE
C       INDICES STORED BY THE VECTORS NROW,NCOL ACCORDING TO THE NEW
C       INDICES STORED BY THE VECTOR NEWNU
C
C       NEL          NUMBER OF NONZERO ELEMENTS OF THE BUS ADMITTANCE
C                    MATRIX
C       N            NUMBER OF BUSES
C
      INTEGER NROW(1),NCOL(1),NEWNU(1),JRYT(1),ICYT(1)
      COMPLEX YT(1),AK(1)
C
      NEL=0
      DO 20 I=1,N
      J1=JRYT(I)
      J2=JRYT(I+1)-1
      DO 10 J=J1,J2
      NEL=NEL+1
      K=ICYT(J)
      NROW(NEL)=NEWNU(I)
      NCOL(NEL)=NEWNU(K)
      AK(NEL)=YT(J)
   10 CONTINUE
   20 CONTINUE
      RETURN
      END
C
C
      SUBROUTINE CHNGC (NB,CVEKT,NEWNUM,CAUX)
C
C       THIS SUBROUTINE CHANGES THE ORDER OF NB ELEMENTS STORED IN
C       THE VECTOR CVEKT ACCORDING TO THE NEW INDICES STORED BY THE
C       VECTOR NEWNUM
C
C       CVEKT        AN AUXILIARY VECTOR
C
      INTEGER NEWNUM(1)
      COMPLEX CAUX(1),CVEKT(1)
C
      DO 10 I=1,NB
```

```
      CAUX(I)=CVEKT(I)
   10 CONTINUE
      DO 20 I=1,NB
      K=NEWNUM(I)
      CVEKT(K)=CAUX(I)
   20 CONTINUE
      RETURN
      END
C
C
      SUBROUTINE CHNGI (N,IVEKT,NEWNUM)
C
C       THIS SUBROUTINE RENUMBERS INDICES STORED IN THE VECTOR IVEK
C       ACCORDING TO THE NEW INDICES STORED BY THE VECTOR NEWNUM
C
      INTEGER NEWNUM(1),IVEKT(1)
C
      DO 10 I=1,N
      IVEKT(I)=NEWNUM(IVEKT(I))
   10 CONTINUE
      RETURN
      END
C
C
      SUBROUTINE CHNGJ (N,IVEKT,NEWNUM,IAUX)
C
C       THIS SUBROUTINE CHANGES THE ORDER OF N ELEMENTS STORED IN
C       THE VECTOR IVEKT ACCORDING TO THE NEW INDICES STORED BY THE
C       VECTOR NEWNUM
C
C       IAUX         AN AUXILIARY VECTOR
C
      INTEGER IVEKT(1),NEWNUM(1),IAUX(1)
C
      DO 10 I=1,N
      IAUX(I)=IVEKT(I)
   10 CONTINUE
      DO 20 I=1,N
      K=NEWNUM(I)
      IVEKT(K)=IAUX(I)
   20 CONTINUE
      RETURN
      END
C
C
      SUBROUTINE CHNGP (NB,BVEKT,NEWNUM,RAUX)
C
C       THIS SUBROUTINE CHANGES THE ORDER OF 2*NB ELEMENTS STORED IN
C       THE VECTOR BVEKT ACCORDING TO THE NEW INDICES STORED BY THE
C       VECTOR NEWNUM
C
C       RAUX         AN AUXILIARY VECTOR
C
      REAL BVEKT(1),RAUX(1)
      INTEGER NEWNUM(1)
C
      J=2*NB
      DO 10 I=1,J
      RAUX(I)=BVEKT(I)
   10 CONTINUE
      DO 20 I=1,NB
      K=NEWNUM(I)
      BVEKT(2*K-1)=RAUX(2*I-1)
      BVEKT(2*K)=RAUX(2*I)
   20 CONTINUE
```

```
      RETURN
      END
C
C
      SUBROUTINE CHNUMB (NB,NINTE,NBOUND,NRET,IBINTE,IBOUND,IBRET,NEWNUM
     1,IOLDNU)
C
C        THIS SUBROUTINE DETERMINES NEW INDICES FOR THE BUSES OF THE
C        POWER SYSTEM. THE NEW INDICES ARE STORED IN THE VECTOR
C        NEWNUM. THE OLD INDICES W.R.T. THE NEW INDICES ARE STORED IN
C        THE VECTOR IOLDNU
C
C        NB             NUMBER OF BUSES
C        NINTE          NUMBER OF INTERNAL BUSES
C        NBOUND         NUMBER OF BOUNDARY BUSES
C        NRET           NUMBER OF RETAINED BUSES
C        IBOUND,IBRET      VECTORS THAT CONTAIN THE CODED INDICES OF BUSES
C                          BELONGING TO THE BOUNDARY AND RETAINED BUSES,
C                          RESPECTIVELY
C
      INTEGER IBINTE(1),IBRET(1),IBOUND(1),NEWNUM(1),IOLDNU(1)
      LOGICAL BELONG
C
      NEXT=NB-NRET-NBOUND-NINTE
      NCEXT=0
      NCRET=NEXT
      NCBOUN=NEXT+NRET
      NCINTE=NCBOUN+NBOUND
C
C        CALCULATE NEW INDICES OF BUSES
C
      DO 40 I=1,NB
      IF (IBINTE(I).EQ.1) GO TO 30
      IF (BELONG(I,IBRET)) GO TO 10
      IF (BELONG(I,IBOUND)) GO TO 20
C
C        CALCULATE A NEW INDEX FOR THE ITH BUS OF THE EXTERNAL SYSTEM
C
      NCEXT=NCEXT+1
      NEWNUM(I)=NCEXT
      IOLDNU(NCEXT)=I
      GO TO 40
   10 CONTINUE
C
C        CALCULATE A NEW INDEX FOR THE RETAINED BUS
C
      NCRET=NCRET+1
      NEWNUM(I)=NCRET
      IOLDNU(NCRET)=I
      GO TO 40
   20 CONTINUE
C
C        CALCULATE A NEW INDEX FOR THE BOUNDARY BUS
C
      NCBOUN=NCBOUN+1
      NEWNUM(I)=NCBOUN
      IOLDNU(NCBOUN)=I
      GO TO 40
   30 CONTINUE
C
C        CALCULUATE A NEW INDEX FOR THE ITH BUS OF THE INTERNAL SYSTEM
C
      NCINTE=NCINTE+1
      NEWNUM(I)=NCINTE
      IOLDNU(NCINTE)=I
```

```
   40 CONTINUE
      RETURN
      END
C
C
      SUBROUTINE EQBCV (N,NLB,NEXT,NFIC,BTYP,BCV,V,NBCV,YT,JRYT,ICYT,OLD
     1NU,IBINTE,NRB,OTPT,IWRITE)
C
C        THIS SUBROUTINE DETERMINES THE NEW VALUES OF BUS CONTROL
C        VARIABLES FOR THE REDUCED SYSTEM
C
C        N              NUMBER OF BUSES
C        NLB            NUMBER OF LOAD BUSES
C        NEXT           NUMBER OF EXTERNAL BUSES
C        NFIC           NUMBER OF FICTITIOUS BUSES
C        JRYT           VECTOR OF LENGTH N+1. IT CONTAINS ROW INDICES OF
C                       A SPARSE BUS ADMITTANCE MATRIX
C        ICYT           VECTOR OF LENGTH JRYT(N+1)-1. IT CONTAINS COLUMN
C                       INDICES OF A SPARSE BUS ADMITTANCE MATRIX
C        YT             A SPARSE BUS ADMITTANCE MATRIX
C        BTYP           VECTOR OF BUS TYPES ( 0 LOAD BUS, 1 GENERATOR BUS,
C                       2 SLACK BUS )
C        V              COMPLEX BUS VOLTAGES (RECTANGULAR COORDINATES)
C        BCV            REAL VECTOR OF LENGTH 2*N. IT CONTAINS NOMINAL
C                       VALUES OF BUS CONTROL VARIABLES
C        NBCV           AUXILIARY VECTOR OF LENGTH 2*N
C        OLDNU          VECTOR OF LENGTH N. IT CONTAINS THE OLD INDICES OF
C                       BUSES W.R.T. NEW INDICES
C        IBINTE         VECTOR OF LENGTH N. IBINTE(K) IS EQUAL TO 1 IF THE
C                       KTH BUS BELONGS TO THE INTERNAL POWER SYSTEM
C        NRB            VECTOR OF LENGTH NLB. IT CONTAINS INDICES OF THE
C                       FICTITIOUS BUSES
C        OTPT           INDEX OF THE OUTPUT UNIT
C        IWRITE         PARAMETER THAT CONTROLS OUTPUT
C
      INTEGER BTYP(1),JRYT(1),ICYT(1),OLDNU(1),IBINTE(1),NRB(1),OTPT
      REAL BCV(1),NBCV(1)
      COMPLEX V(1),YT(1),YI,EQCURR
C
      NEX=N-NEXT
C
C        MODIFICATION OF THE VECTOR V OF THE INITIAL BUS VOLTAGES
C
      DO 10 I=1,NEX
      J=NEXT+I
      V(I)=V(J)
   10 CONTINUE
C
      IF (NFIC.EQ.0) GO TO 30
C
C        DETERMINE THE INITIAL BUS VOLTAGES FOR THE FICTITIOUS BUSES
C
      K=NEX+NFIC
      V(K)=V(NEX)
      DO 20 I=1,NFIC
      K=NEX+I-1
      J=NRB(I)
      V(K)=V(J)
   20 CONTINUE
C
   30 CONTINUE
C
C        CALCULATE AND STORE THE NEW BUS CONTROL VARIABLES IN THE VECTOR
C        NBCV
C
```

```
      K=NEX-1
      IF (NFIC.EQ.0) K=NEX
      DO 60 I=1,K
      IK=I+NEXT
      IF (BTYP(IK).EQ.2) GO TO 60
      J=OLDNU(IK)
      IF (IBINTE(J).EQ.1) GO TO 50
C
C        CALCULATE THE CURRENT INJECTED IN THE ITH BUS
C
      YI=EQCURR(V,YT,JRYT,ICYT,I)
C
C        REAL POWER
C
      NBCV(2*I-1)=REAL(V(I)*CONJG(YI))
      IF (BTYP(IK).NE.0) GO TO 40
C
C        REACTIVE POWER
C
      NBCV(2*I)=AIMAG(V(I)*CONJG(YI))
      GO TO 60
   40 CONTINUE
C
C        MODULUS OF THE BUS VOLTAGE
C
      NBCV(2*I)=BCV(2*IK)
      GO TO 60
   50 CONTINUE
      NBCV(2*I-1)=BCV(2*IK-1)
      NBCV(2*I)=BCV(IK*2)
   60 CONTINUE
      IF (BTYP(N).NE.2) GO TO 70
      NBCV(2*NEX-1)=BCV(2*N-1)
      NBCV(2*NEX)=BCV(2*N)
      GO TO 80
   70 CONTINUE
      IF (NFIC.EQ.0) GO TO 80
      K=NEX+NFIC
C
C        CALCULATE THE CURRENT INJECTED IN THE KTH BUS
C
      YI=EQCURR(V,YT,JRYT,ICYT,K)
C
C        REAL POWER
C
      NBCV(2*NEX-1)=REAL(V(K)*CONJG(YI))
C
C        MODULUS OF THE BUS VOLTAGE
C
      NBCV(2*NEX)=CABS(V(K))
   80 CONTINUE
C
C        THE NEW BUS CONTROL VARIABLES ARE STORED IN THE MATRIX BCV
C
      DO 90 I=1,NEX
      BTYP(I)=BTYP(I+NEXT)
      BCV(2*I-1)=NBCV(2*I-1)
      BCV(2*I)=NBCV(2*I)
   90 CONTINUE
C
C        DETERMINE THE BUS CONTROL VARIABLES FOR THE FICTITIOUS BUSES
C
      IF (NFIC.EQ.0) GO TO 110
      K=NEX+NFIC
      BTYP(K)=BTYP(NEX)
```

```
      BCV(2*K-1)=BCV(2*NEX-1)
      BCV(2*K)=BCV(2*NEX)
      DO 100 I=1,NFIC
      K=NEX+I-1
      J=NRB(I)
      BTYP(K)=1
      BCV(2*K-1)=0.0
      BCV(2*K)=CABS(V(J))
  100 CONTINUE
  110 CONTINUE
C
C        CALCULATE THE NUMBER OF THE BUSES N AND THE NUMBER OF THE LOAD
C        BUSES NLB OF THE REDUCED SYSTEM
C
      N=NEX+NFIC
      NLB=0
      DO 120 I=1,N
      IF (BTYP(I).NE.0) GO TO 120
      NLB=NLB+1
  120 CONTINUE
      NB=N
C
      IF (IWRITE.LT.3) RETURN
C
C        PRINT OUT THE RESULTS
C
      WRITE (OTPT,150)
      DO 140 I=1,N
      IF (I.GT.NEX-1.AND.I.LT.N) GO TO 130
      K=NEXT+I
      IF (I.EQ.N) K=K-NFIC
      J=OLDNU(K)
      WRITE (OTPT,160) J,BTYP(I),V(I),BCV(2*I-1),BCV(2*I)
      GO TO 140
  130 CONTINUE
      WRITE (OTPT,170) I,BTYP(I),V(I),BCV(2*I-1),BCV(2*I)
  140 CONTINUE
  150 FORMAT (1H /,43X,"BASE CASE"/,21X,"VECTOR OF BUS VOLTAGES"//,3X,"B
     1US",2X,"BTYP",7X," RECTANGULAR COORDINATES",7X,"VECTOR OF BUS CONT
     2ROL VARIABLES"/)
  160 FORMAT (2X,2I4,6X,2(E13.5,2X),2(4X,E13.5,2X))
  170 FORMAT (1X,"F",2I4,6X,2(E13.5,2X),2(4X,E13.5,2X))
      RETURN
      END
C
C
      SUBROUTINE EQCCC (NLIN,NTL,LBINP,LBOUT,LINPG,LINPB,LG,LB,LOUTG,LOU
     1TB,LTAP,JRYT,ICYT,YT,NB,NFIC,NEXT,NEL,IFLAG)
C
C        THIS SUBROUTINE REMOVES THE NLINTH LINE FROM THE POWER SYSTEM
C        DESCRIBED BY AN ADMITTANCE MATRIX  (VECTORS YT,JRYT,ICYT) AND
C        LINE PARAMETERS ( MATRICES LBINP,LBOUT,LINPG,LINPB,LB,LG,LOUTG,
C        LOUTB,LTAP)
C
      INTEGER LBINP(1),LBOUT(1),JRYT(1),ICYT(1)
      REAL LINPG(1),LINPB(1),LB(1),LG(1),LOUTG(1),LOUTB(1),LTAP(1)
      COMPLEX YT(1),Y,D1,D2
C
      IFLAG=0
      IF (NLIN.EQ.0) RETURN
      IF (NLIN.GT.NTL) GO TO 10
      R=1.0
      R1=0.0
      KB=NB-NFIC
C
```

```
C        DETERMINE INDICES OF BUSES INCIDENT WITH THE NLINTH LINE
C
      I=LBINP(NLIN)
      J=LBOUT(NLIN)
      I1=I-NEXT
      J1=J-NEXT
      IF (I1.EQ.KB) I1=NB
      IF (J1.EQ.KB) J1=NB
C
C        DETERMINE A POSITION OF INDICES (I1,J1) AND (J1,I1) IN THE
C        MATRIX ICYT
C
      CALL EQDDD (I1,J1,JRYT,ICYT,IY1,JY1,IFLAG)
      IF (IFLAG.LT.0) RETURN
      Y=CMPLX(LG(NLIN),LB(NLIN))
      D1=CMPLX(LINPG(NLIN),LINPB(NLIN))
      D2=CMPLX(LOUTG(NLIN),LOUTB(NLIN))
      ID=JRYT(I1)
      JD=JRYT(J1)
C
C        CHANGE VALUES OF ELEMENTS OF THE ADMITTANCE MATRIX
C
      YT(ID)=YT(ID)-R*(Y/(LTAP(NLIN)**2)+D1)
      YT(JD)=YT(JD)-R*(Y+D2)
      YT(IY1)=-R1*Y/LTAP(NLIN)
      YT(JY1)=-R1*Y
C
C        MODIFY THE VECTORS YT,JRYT,ICYT
C
      IF (IY1.GT.JY1) CALL EQREEY (I1,IY1,NB,NEL,JRYT,ICYT,YT)
      CALL EQREEY (J1,JY1,NB,NEL,JRYT,ICYT,YT)
      IF (JY1.GT.IY1) CALL EQREEY (I1,IY1,NB,NEL,JRYT,ICYT,YT)
      RETURN
   10 CONTINUE
      IFLAG=-10
      WRITE (6,20) IFLAG
   20 FORMAT (2X,"IFLAG FROM EQCCC IFLAG= ",I3,3X,"THE INDEX OF THE LINE
     1 IS GREATER THAN THE NUMBER OF ALL LINES")
      RETURN
      END
C
C
      SUBROUTINE EQCOMP (LBINP,LBOUT,LINPG,LINPB,LG,LB,LOUTG,LOUTB,LTAP,
     1BV,V,VCB,NB,NTL,NLIN,MODE3,NBOUND,JBOUND,NRET,JRET,JBINTE,JOLDNU,N
     2EXT,NFIC,ICONT,INPT,OTPT,IFLAG)
C
C        SUBROUTINE EQCOMP CALCULATES AND COMPARES POWER FLOWS IN
C        TRANSMISSION LINES OF THE UNREDUCED POWER SYSTEM AND THE REDUCED
C        POWER SYSTEM UNDER A CONTINGENCY
C
C        ALL THE ARGUMENTS OF THIS SUBROUTINE ARE DESCRIBED IN THE
C        SUBROUTINE EQCONTI
C
      INTEGER LBINP(1),LBOUT(1),JBOUND(1),JRET(1),JOLDNU(1),JBINTE(1),OT
     1PT
      REAL LINPG(1),LINPB(1),LG(1),LB(1),LOUTG(1),LOUTB(1),LTAP(1)
      COMPLEX V(1),VCB(1),V1,V2,V3,V4,DV1,DV2,Y,C,C1,CNI,F,F1,FIN,PIN,PI
     1NF,BV(1)
C
C        READ A LOAD FLOW SOLUTION OF THE UNREDUCED SYSTEM WITH THE
C        NLINTH LINE REMOVED
C
      IKLIN=0
   10 CONTINUE
      READ (INPT,*) NULIN
```

```
      IKLIN=IKLIN+1
      IF (IKLIN.GT.NTL) GO TO 200
      DO 20 I=1,NB
      READ (INPT,*) K,R1,R2,R3,R4
      VCB(I)=CMPLX(R1,R2)
   20 CONTINUE
      IF (NULIN.NE.NLIN) GO TO 10
C
C        END OF READING
C
      WRITE (OTPT,250)
C
C        PRINT OUT THE TITLE
C
      NBI=NB-NEXT
      GO TO (30,40,50,60,70), MODE3
   30 WRITE (OTPT,280) NBI
      GO TO 80
   40 WRITE (OTPT,290) NBI
      GO TO 80
   50 WRITE (OTPT,300) NBI
      GO TO 80
   60 WRITE (OTPT,310) NBI
      GO TO 80
   70 CONTINUE
      WRITE (OTPT,320) NBI
   80 CONTINUE
      WRITE (OTPT,260) (JBOUND(I),I=1,NBOUND)
      DO 90 I=1,NBOUND
      JBINTE(JBOUND(I))=1
   90 CONTINUE
      IF (NRET.EQ.0) GO TO 110
      WRITE (OTPT,270) (JRET(I),I=1,NRET)
      DO 100 I=1,NRET
      JBINTE(JRET(I))=1
  100 CONTINUE
  110 CONTINUE
      IF (NLIN.NE.0) GO TO 120
      WRITE (OTPT,340)
      GO TO 130
  120 CONTINUE
      K1=LBINP(NLIN)
      K2=LBOUT(NLIN)
      WRITE (OTPT,330) NLIN,K1,K2
  130 CONTINUE
      WRITE (OTPT,380)
      WRITE (OTPT,240)
      WRITE (OTPT,350)
      WRITE (OTPT,360)
C
C        COMPUTE AND COMPARE THE POWER FLOW IN ALL LINES FOR THE
C        UNREDUCED AND THE REDUCED POWER SYSTEM
C
      DO 140 I=1,NTL
C
C        I IS THE INDEX OF A LINE
C
      IF (I.EQ.NLIN) GO TO 140
      K1=LBINP(I)
      K2=LBOUT(I)
      IF (JBINTE(K1).EQ.0.OR.JBINTE(K2).EQ.0) GO TO 140
C
C        COMPUTE THE CURRENT FLOWING IN ITH LINE OF THE REDUCED SYSTEM
C
      V1=V(K1)/(LTAP(I))**2
```

```
      V2=V(K2)/LTAP(I)
      DV1=V1-V2
      Y=CMPLX(LG(I),LB(I))
      C=DV1*Y
      C1=V(K1)*CMPLX(LINPG(I),LINPB(I))
C
C       THE CURRENT FLOWING IN THE ITH LINE OF THE REDUCED SYSTEM
C
      CNI=C+C1
C
C       COMPUTE THE CURRENT FLOWING IN THE ITH LINE OF THE UNREDUCED
C       SYSTEM
C
      V3=VCB(K1)/(LTAP(I))**2
      V4=VCB(K2)/LTAP(I)
      DV2=V3-V4
      F=DV2*Y
      F1=VCB(K1)*CMPLX(LINPG(I),LINPB(I))
C
C       THE CURRENT FLOWING IN THE ITH LINE OF THE UNREDUCED SYSTEM
C
      FIN=F+F1
C
C       THE POWER INJECTED IN THE ITH LINE OF THE REDUCED SYSTEM
C
      PIN=V(K1)*CONJG(CNI)
      PINR=REAL(PIN)
      PINIM=AIMAG(PIN)
C
C       THE POWER INJECTED IN THE ITH LINE OF THE UNREDUCED SYSTEM
C
      PINF=VCB(K1)*CONJG(FIN)
      PINFR=REAL(PINF)
      PINFIM=AIMAG(PINF)
C
C       COMPARE RESULTS
C
      X1=PINFR-PINR
      X2=PINFIM-PINIM
      X3=(100.0*X1)/PINFR
      X4=(100.0*X2)/PINFIM
      X5=(100.0*X1)/CABS(PINF)
      X7=(100.0*CABS(PINF-PIN))/CABS(PINF)
      WRITE (OTPT,370) I,K1,K2,PINFR,PINFIM,PINR,PINIM,X1,X2,X3,X4,X7
  140 CONTINUE
C
C       COMPUTE AND COMPARE THE CHANGES OF THE BUS VOLTAGES
C       CORRESPONDING TO A CHANGE IN OPERATING CONDITIONS IN THE
C       INTERNAL PART OF THE UNREDUCED AND REDUCED POWER SYSTEMS
C
      WRITE (OTPT,390)
      WRITE (OTPT,420)
C
      DO 160 I=1,NB
      IF (JBINTE(I).EQ.0) GO TO 160
      C=VCB(I)-V(I)
      X1=REAL(VCB(I))
      X2=AIMAG(VCB(I))/X1
      X3=REAL(V(I))
      X4=AIMAG(V(I))/X3
      X5=ATAN(X2)
      X6=ATAN(X4)
      X6=X5-X6
      IF (X5.NE.0.0) X7=100.0*X6/X5
      X=CABS(C)
```

```
      IF (X5.NE.0.0) GO TO 150
      WRITE (OTPT,440) I,X,X6
      GO TO 160
  150 CONTINUE
      WRITE (OTPT,440) I,X,X6,X7
  160 CONTINUE
      IF (ICONT.GT.1) RETURN
C
C        COMPARE THE CHANGES OF THE BUS VOLTAGES CORRESPONDING TO A
C        CHANGE IN OPERATING CONDITIONS IN THE INTERNAL PART OF THE
C        UNREDUCED SYSTEM WITH THE LOAD FLOW SOLUTION FOR THE BASE CASE
C
      WRITE (OTPT,400)
      WRITE (OTPT,430)
C
      DO 180 I=1,NB
      IF (JBINTE(I).EQ.0) GO TO 180
      C=VCB(I)-BV(I)
      X1=REAL(VCB(I))
      X2=AIMAG(VCB(I))/X1
      X3=REAL(BV(I))
      X4=AIMAG(BV(I))/X3
      X5=ATAN(X2)
      X6=ATAN(X4)
      X6=X5-X6
      IF (X5.NE.0.0) X7=100.0*X6/X5
      X=CABS(C)
      IF (X5.NE.0.0) GO TO 170
      WRITE (OTPT,440) I,X,X6
      GO TO 180
  170 CONTINUE
      WRITE (OTPT,440) I,X,X6,X7
  180 CONTINUE
C
C        COMPUTE THE POWER FLOW IN THE LINES FOR THE BASE CASE
C
      WRITE (OTPT,410)
      WRITE (OTPT,230)
C
      DO 190 I=1,NTL
      K1=LBINP(I)
      K2=LBOUT(I)
      IF (JBINTE(K1).EQ.0.OR.JBINTE(K2).EQ.0) GO TO 190
C
C        COMPUTE THE CURRENT FLOWING IN THE ITH LINE
C
      V1=BV(K1)/(LTAP(I))**2
      V2=BV(K2)/LTAP(I)
      DV1=V1-V2
      Y=CMPLX(LG(I),LB(I))
      C=DV1*Y
      C1=BV(K1)*CMPLX(LINPG(I),LINPB(I))
C
C        THE CURRENT FLOWING IN THE ITH LINE
C
      CNI=C+C1
C
C        THE POWER INJECTED IN THE ITH LINE
C
      PIN=BV(K1)*CONJG(CNI)
      PINR=REAL(PIN)
      PINIM=AIMAG(PIN)
      WRITE (OTPT,220) I,K1,K2,PINR,PINIM
  190 CONTINUE
      RETURN
```

```
  200 CONTINUE
C
C        THE LACK OF THE LOAD FLOW SOLUTION FOR THE LINE WHICH IS
C        DECLARED
C
      IFLAG=-10
      WRITE (OTPT,210) IFLAG
      RETURN
  210 FORMAT (2X,"RETURN FLAG FROM EQCOMP IFLAG= ",I3)
  220 FORMAT (2X,3(I4,2X),2(F8.3,3X))
  230 FORMAT (//,2X,"LINE        LINE   ",7X," POWER FLOW   ",/,4X,"NO",2X,"
     1FROM",4X,"TO",7X,"P",10X,"Q")
  240 FORMAT (2X,"P - REAL POWER   ;   Q - REACTIVE POWER   ; S - COMPLEX P
     1OWER")
  250 FORMAT (///,2X,"RESULTS OF CONTINGENCY ANALYSIS:TEST USING EXTERNA
     1L EQUIVALENTS",//)
  260 FORMAT (//,2X,"BOUNDARY BUSES: ",20(I4,1X))
  270 FORMAT (/,2X,"RETAINED BUSES:",20(I4,1X))
  280 FORMAT (2X,"THE SOLUTION FOR ",I4,1X,"BUS-SYSTEM"," USING STANDARD
     1 WARD METHOD")
  290 FORMAT (2X,"THE SOLUTION FOR ",I4,1X,"BUS-SYSTEM"," USING STANDARD
     1 WARD METHOD WITH BUFFER ZONE")
  300 FORMAT (2X,"THE SOLUTION FOR ",I4,1X,"BUS-SYSTEM"," USING EXTENDED
     1 WARD METHOD")
  310 FORMAT (2X,"THE SOLUTION FOR ",I4,1X,"BUS-SYSTEM"," USING SIMPLIFI
     1ED EXTENDED WARD METHOD")
  320 FORMAT (2X,"THE SOLUTION FOR ",I4,1X,"BUS-SYSTEM"," USING STANDARD
     1 WARD METHOD WITH SHUNTS IN THE EXTERNAL SYSTEM")
  330 FORMAT (//,2X,"OUTAGE LINE: ",I4,2X,"FROM: ",I4,"  TO: ",I4)
  340 FORMAT (//,2X,"THE SOLUTION FOR THE BASE CASE")
  350 FORMAT (//,2X,"LINE       LINE   ",3X,"UNREDUCED SYSTEM",3X,"REDUCED
     1 SYSTEM",10X,"DEVIATIONS",6X,"(DEVIA.P)/P",6X,"(DEVIA.Q)/Q",6X,"MO
     2D(DEVIA.S)/MOD(S)")
  360 FORMAT (4X,"NO",2X,"FROM",4X,"TO",6X," P",7X," Q   ",6X,"P ",5X,"
     1Q",10X," P",6X,"  Q",7X,"ERROR IN %",7X,"ERROR IN %",8X,"ERROR IN
     2%")
  370 FORMAT (2X,3(I4,2X),4(F7.3,2X),3X,2(F7.3,2X),6X,3(F7.2,10X))
  380 FORMAT (/,2X,"THE POWER FLOW IN LINES CORRESPONDING TO A CHANGE IN
     1 OPERATING CONDITIONS IN THE INTERNAL "/,2X,"PART OF THE UNREDUCED
     2 AND REDUCED SYSTEM")
  390 FORMAT ("1",1X,"THE CHANGES OF THE BUS VOLTAGES CORRESPONDING TO A
     1 CHANGE IN OPERATING CONDITIONS IN THE INTERNAL "/,2X,"PART OF THE
     2 UNREDUCED AND REDUCED SYSTEM")
  400 FORMAT ("1",1X,"THE CHANGES OF THE BUS VOLTAGES CORRESPONDING TO A
     1 CHANGE IN OPERATING CONDITIONS IN THE INTERNAL "/,2X,"PART OF THE
     2  UNREDUCED SYSTEM W.R.T SOLUTION FOR THE BASE CASE")
  410 FORMAT (/,2X,"THE POWER FLOW IN TRANSMISSION LINES FOR THE BASE CA
     1SE")
  420 FORMAT (//1X,"MODULUS OF THE BUS VOLTAGE ERROR",2X,//4X,"BUS",6X,"
     1VALUE",23X,"DEVIATION OF ANGLE",3X,"ANGLE ERROR IN %")
  430 FORMAT (//1X,"MODULUS OF THE CHANGE OF THE BUS VOLTAGE ",2X,//4X,"
     1BUS",6X,"VALUE",23X,"THE CHANGE OF ANGLE",3X,"THE CHANGE OF ANGLE
     2IN %")
  440 FORMAT (3X,I4,3X,F8.4,32X,2(F7.3,12X))
      END
C
C
      SUBROUTINE EQCONTI (INB,NB,NLB,NEXT,NFIC,NBOUND,JBOUND,NRET,JRET,B
     1TYP,BCV,V,NEWNUM,JOLDNU,JBINTE,YT,JRYT,ICYT,LBINP,LBOUT,NTL,MODE3,
     2NEL,NLIN,LINPG,LINPB,LG,LB,LOUTG,LOUTB,LTAP,BV,ITEL,TOLV,T,WSP,LWS
     3P,ICONT,IFLAG,INPT,OTPT,IFOREQ,IWRITE,IPRINT)
C
C        SUBROUTINE EQCONTI PERFORMS A CONTINGENCY ANALYSIS OF THE
C        REDUCED POWER SYSTEM
C
```

```
C       INB            NUMBER OF BUSES IN THE UNREDUCED POWER SYSTEM
C       NBEQ           NUMBER OF BUSES IN THE REDUCED POWER SYSTEM
C       NLBE           NUMBER OF LOAD BUSES IN THE REDUCED POWER SYSTEM
C       NEXT           NUMBER OF BUSES BELONGING TO THE EXTERNAL POWER
C                      SYSTEM
C       NFIC           NUMBER OF FICTITIOUS BUSES
C       NBOUND         NUMBER OF BOUNDARY BUSES
C       JBOUND         VECTOR OF LENGTH NBOUND. IT CONTAINS BOUNDARY BUSES
C       NRET           NUMBER OF RETAINED BUSES
C       JRET           VECTOR OF LENGTH NRET. IT CONTAINS RETAINED BUSES
C       BTYP           VECTOR OF BUS TYPES ( 0 LOAD BUS, 1 GENERATOR BUS,
C                      2 SLACK BUS ) OF THE REDUCED POWER SYSTEM
C       BCV            VECTOR OF LENGTH 2*NB. IT CONTAINS NOMINAL VALUES
C                      OF BUS CONTROL VARIABLES OF THE REDUCED POWER
C                      SYSTEM
C       V              COMPLEX BUS VOLTAGES (RECTANGULAR COORDINATES) OF THE
C                      REDUCED POWER SYSTEM
C       NEWNUM         VECTOR OF LENGTH NB. IT STORES THE NEW INDICES OF
C                      BUSES
C       IOLDNU         VECTOR OF LENGTH NB. IT STORES THE OLD INDICES OF
C                      BUSES W.R.T. NEW INDICES
C       IBINTE         VECTOR OF LENGTH NB. IT STORES BUSES OF THE INTERNAL
C                      POWER SYSTEM
C       YT             THE SPARSE BUS ADMITTANCE MATRIX
C       JRYT           VECTOR OF LENGTH NB+1. IT CONTAINS ROW INDICES OF
C                      THE SPARSE BUS ADMITTANCE MATRIX
C       ICYT           VECTOR OF LENGTH JRYT(NB+1)-1. IT CONTAINS COLUMN
C                      INDICES OF THE SPARSE ADMITTANCE MATRIX
C       LBINP,LBOUT         VECTORS OF LENGTH NTL. LBINP(K),LBOUT(K) CONTAIN
C                           INDICES OF BUSES INCIDENT WITH THE KTH LINE
C       NTL            NUMBER OF TRANSMISSION LINES
C       MODE3          TYPE OF AN EQUIVALENT :
C                      1 - STANDARD WARD EQUIVALENT WITHOUT SHUNTS
C                      2 - STANDARD WARD EQUIVALENT WITH BUFFER ZONE
C                      3 - EXTENDED WARD EQUIVALENT
C                      4 - SIMPLIFIED EXTENDED WARD EQUIVALENT
C                      5 - STANDARD WARD EQUIVALENT WITH SHUNTS IN THE
C                          EXTERNAL SYSTEM
C       NEL            NUMBER OF NONZERO ELEMENTS OF THE BUS ADMITTANCE
C                      MATRIX
C       NLIN           THE INDEX OF A LINE TO BE REMOVED
C       LINPG,LINPB          VECTORS OF LENGTH NTL. LINPG(K),LINPB(K) CONTAIN
C                            THE INPUT SHUNT CONDUCTANCE AND SUSCEPTANCE OF
C                            THE KTH LINE
C       LG,LB          VECTORS OF LENGTH NTL. LG(K),LB(K) CONTAIN THE
C                      CONDUCTANCE AND SUSCEPTANCE OF THE KTH TRANSMISSION
C                      LINE
C       LOUTG,LOUTB          VECTORS OF LENGTH NTL. LOUTG(K),LOUTB(K) CONTAIN
C                            THE OUTPUT SHUNT CONDUCTANCE AND SUSCEPTANCE OF
C                            THE KTH TRANSMISSION LINE
C       LTAP           VECTOR OF LENGTH NTL. LTAP(K) CONTAINS THE KTH LINE
C                      TRANSFORMER RATIO
C       BV             COMPLEX VECTOR. IT CONTAINS THE VALUES OF THE BUS
C                      VOLTAGES ( IN RECTANGULAR COORDINATES) OF THE
C                      UNREDUCED SYSTEM
C       ITEL           THE NUMBER OF ITERATIONS
C       TOLV           REQUIRED ACCURACY OF THE LOAD FLOW SOLUTION
C       T              TOTAL ITERATION TIME
C       WSP            REAL VECTOR. WORKSPACE
C       LWSP           THE LENGTH OF THE WORKSPACE
C       ICONT          IT CONTROLS THE MODE OF A WORK
C                      0  ONLY LOAD FLOW SOLUTION IS PERFORMED
C                      1  THE BUS VOLTAGES OF THE UNREDUCED SYSTEM ,BEFORE
C                         AND AFTER OUTAGE ARE COMPARED
C                      2  ONLY POWER FLOW IN LINES BEFORE AND AFTER OUTAGE
```

```
C                   IS COMPARED
C       IFLAG       RETURN FLAG FROM SUBROUTINE (SEE.REPORT SOS-83-24-U)
C       INPT        INDEX OF THE INPUT UNIT. IF (ICONT.NE.0) THE DATA
C                   DESCRIBING THE UNREDUCED POWER SYSTEM IS READ FROM
C                   THE UNIT INPT
C       OTPT        INDEX OF THE OUTPUT UNIT
C       IFOREQ      INDEX OF THE INPUT UNIT. IF (IFOREQ>0) THE DATA
C                   DESCRIBING THE REDUCED POWER SYSTEM IS READ FROM THE
C                   UNIT IFOREQ
C       IWRITE      PARAMETER THAT CONTROLS OUTPUT (SEE.REP. SOS-83-24-U)
C       IPRINT      PARAMETER THAT CONTROLS OUTPUT. IF IPRINT=1 AND
C                   IFOREQ>0 THE DESCRIPTION OF THE REDUCED POWER SYSTEM
C                   IS PRINTED OUT ON THE UNIT OTPT
C
      INTEGER BTYP(1),JRYT(1),ICYT(1),LBINP(1),LBOUT(1),OTPT,JBOUND(1),J
     1RET(1),NEWNUM(1),JOLDNU(1),JBINTE(1)
      REAL WSP(1),BCV(1),LINPG(1),LINPB(1),LG(1),LB(1),LOUTG(1),LOUTB(1)
     1,LTAP(1)
      COMPLEX YT(1),V(1),BV(1)
C
      IFLAG=0
      WRITE (OTPT,40)
      IF (IFOREQ.LT.0) GO TO 10
C
C       READ THE DATA DESCRIBING THE REDUCED SYSTEM
C
      CALL EQREAD (INB,NB,NLB,NEXT,NFIC,NBOUND,JBOUND,NRET,JRET,BTYP,BCV
     1,V,NEWNUM,JOLDNU,JBINTE,YT,JRYT,ICYT,LBINP,LBOUT,NTL,MODE3,IFOREQ,
     2OTPT,IPRINT)
      NEL=JRYT(NB+1)-1
C
   10 CONTINUE
C
C       THE NLINTH LINE WILL BE REMOVED
C
C
      CALL EQCCC (NLIN,NTL,LBINP,LBOUT,LINPG,LINPB,LG,LB,LOUTG,LOUTB,LTA
     1P,JRYT,ICYT,YT,NB,NFIC,NEXT,NEL,IFLAG)
      IF (IFLAG.LT.0) RETURN
      MODE=0
C
C       DETERMINE A LOAD FLOW SOLUTION
C       SUBROUTINE LFLFD1M IS CALLED FROM THE TTM1 LIBRARY
C
      CALL LFLFD1M (NB,NLB,NEL,JRYT,ICYT,BTYP,YT,V,BCV,WSP,LWSP,ITEL,TOL
     1V,T,MODE,IFLAG,OTPT,IWRITE)
      IF (IFLAG.LT.0.OR.ICONT.EQ.0) RETURN
C
C       PREPARE DATA FOR THE SUBROUTINE EQCOMP
C
      CALL EQMOSO (INB,NB,NFIC,NEXT,V,WSP,JOLDNU,MODE3)
C
C       RENUMBER INDICES STORED IN THE VECTORS LBINP,LBOUT
C
      DO 20 I=1,NTL
      LBINP(I)=JOLDNU(LBINP(I))
      LBOUT(I)=JOLDNU(LBOUT(I))
   20 CONTINUE
      IFLAG1=0
C
C       PROCESS CONTINGENCY SOLUTIONS FOR THE UNREDUCED AND REDUCED
C       POWER SYSTEM
C
      CALL EQCOMP (LBINP,LBOUT,LINPG,LINPB,LG,LB,LOUTG,LOUTB,LTAP,BV,V,W
     1SP,INB,NTL,NLIN,MODE3,NBOUND,JBOUND,NRET,JRET,JBINTE,JOLDNU,NEXT,N
```

```
     2FIC,ICONT,INPT,OTPT,IFLAG1)
      CALL CHNGI (NTL,LBINP,NEWNUM)
      CALL CHNGI (NTL,LBINP,NEWNUM)
      IF (IFLAG1.LT.0) WRITE (OTPT,30) IFLAG1
   30 FORMAT (2X,"IFLAG FROM EQCOMP= ",I3)
   40 FORMAT (2X,"CONTINGENCY ANALYSIS OF THE REDUCED POWER SYSTEM"/)
      RETURN
      END
C
C
      COMPLEX FUNCTIONEQCURR(X,YT,JRYT,ICYT,MB)
C
C       FUNCTION SUBPROGRAM EQCURR CALCULATES THE VALUE OF THE
C       CURRENT INJECTED INTO THE MBTH BUS FOR GIVEN VECTOR X
C       OF BUS VOLTAGES
C
      INTEGER JRYT(1),ICYT(1)
      COMPLEX X(1),YT(1)
C
      K1=JRYT(MB)
      K2=JRYT(MB+1)-1
      EQCURR=(0.0,0.0)
      DO 10 I=K1,K2
      EQCURR=EQCURR+YT(I)*X(ICYT(I))
   10 CONTINUE
      RETURN
      END
C
C
      SUBROUTINE EQDDD (I,J,IRYT,ICYT,I1,J1,IFLAG)
C
C       THIS SUBROUTINE DETERMINES FOR INDICES (I,J) A POSITION OF
C       THE JTH INDEX IN THE VECTOR ICYT AND FOR INDICES (J,I) A
C       POSITION OF THE ITH INDEX IN THE VECTOR ICYT
C
      INTEGER IRYT(1),ICYT(1)
C
      IFLAG=0
      K1=IRYT(I)+1
C
C       CALCULATE THE POSITION OF THE JTH INDEX FOR THE PAIR (I,J)
C
      K2=IRYT(I+1)-1
      DO 10 K3=K1,K2
      IF (ICYT(K3).NE.J) GO TO 10
      I1=K3
      GO TO 20
   10 CONTINUE
      IFLAG=-10
      WRITE (6,50) IFLAG
      RETURN
   20 CONTINUE
C
C       THE ELEMENT J IS IN THE POSITION I1
C
C       CALCULATE THE POSITION OF THE ITH INDEX FOR THE PAIR (J,I)
C
      K1=IRYT(J)+1
      K2=IRYT(J+1)-1
      DO 30 K3=K1,K2
      IF (ICYT(K3).NE.I) GO TO 30
      J1=K3
      GO TO 40
   30 CONTINUE
      IFLAG=-11
```

```
      WRITE (6,50)
      RETURN
   40 CONTINUE
C
C         THE ELEMENT I IS IN THE POSITION J1
C
      RETURN
   50 FORMAT (2X,"IFLAG FROM EQDDD IFLAG=  ",I3,3X,"AN ERROR IN THE MATRI
     1X ICYT")
      END
```

```
C
C
      SUBROUTINE EQDEIE (NB,NTL,LBINP,LBOUT,NBOUND,JBOUND,NRET,JRET,NEWN
     1UM,IOLDNU,IBINTE,NEXT,WSP,LWSP,MODE,OTPT,IFLAG,IWRITE)
C
C        THIS SUBROUTINE DIVIDES A SET OF POWER SYSTEM BUSES INTO
C        INTERNAL AND EXTERNAL BUSES, DETERMINES NEW INDICES OF THE
C        BUSES AND PARTLY PERFORMS RENUMBERING
C
C        NB            NUMBER OF BUSES
C        NTL           NUMBER OF TRANSMISSION LINES
C        LBINP,LBOUT       VECTORS OF LENGTH NTL. LBINP(K),LBOUT(K) CONTAIN
C                          INDICES OF BUSES INCIDENT WITH THE KTH LINE
C        NBOUND        NUMBER OF BOUNDARY BUSES
C        JBOUND        VECTOR OF LENGTH NBOUND. IT CONTAINS BOUNDARY BUSES
C        NRET          NUMBER OF RETAINED BUSES
C        JRET          VECTOR OF LENGTH NRET. IT CONTAINS RETAINED BUSES
C        NEWNUM        VECTOR OF LENGTH NB. IT STORES THE NEW INDICES OF
C                      BUSES
C        IOLDNU        VECTOR OF LENGTH NB. IT STORES THE OLD INDICES OF
C                      BUSES W.R.T. THE NEW INDICES
C        IBINTE        VECTOR OF LENGTH NB. IBINTE(K) IS SET TO 1 IF THE KTH
C                      BUS BELONGS TO THE INTERNAL POWER SYSTEM
C        NEXT          NUMBER OF BUSES BELONGING TO THE EXTERNAL POWER
C                      SYSTEM
C        WSP           REAL WORKSPACE
C        LWSP          LENGTH OF THE WORKSPACE WSP
C        MODE          MODE OF AN OPERATION
C                      0 - A REDUCED POWER SYSTEM IS DETERMINED
C                      1 - AN EXTERNAL EQUIVALENT IS DETERMINED
C        OTPT          INDEX OF THE OUTPUT UNIT
C        IFLAG         RETURN FLAG:
C                       0 - NORMAL RETURN
C                      -2 - THE LENGTH OF THE WORKSPACE IS TOO SMALL
C        IWRITE        PARAMETER THAT CONTROLS OUTPUT
C
      INTEGER LBINP(1),LBOUT(1),JBOUND(1),JRET(1),NEWNUM(1),IOLDNU(1),IB
     1INTE(1),WSP(1),OTPT
      COMPLEX V(1)
C
      IF (NBOUND.EQ.0.AND.NRET.EQ.0) RETURN
      CALL SECOND (TTM1)
C
C        DISTRIBUTE THE WORKSPACE
C
      K=(NB-1)/50+1
      JBSTOS=1
      JBOUN=JBSTOS
      JJBRET=JBSTOS+K
      IAK=JJBRET+K+1
      ITS=IAK+NB+2
      IAS=ITS+2*NTL+1
      IF (LWSP.GE.IAS+NB) GO TO 10
      IFLAG=-2
      RETURN
   10 CONTINUE
      NINTE=0
C
C        SET TO ZERO THE VECTORS JBOUN,JJBRET AND IBINTE
C
      CALL ZERO (K,WSP(JBOUN))
      CALL ZERO (K,WSP(JJBRET))
      CALL ZERO (NB,IBINTE)
      NINTE=0
C
```

```
C        CALCULATE THE ADJACENCY MATRIX
C
      IF (MODE.EQ.0) CALL ORDEVE (NB,NTL,LBINP,LBOUT,WSP(IAK),WSP(ITS))
C
C        THE BOUNDARY AND RETAINED BUSES ARE CODED
C
      IF (NBOUND.NE.0) CALL SETBIN (NBOUND,JBOUND,WSP(JBOUN))
      IF (NRET.NE.0) CALL SETBIN (NRET,JRET,WSP(JJBRET))
C
C        DETERMINE THE BUSES BELONGING TO THE INTERNAL
C        POWER SYSTEM
C
      WSP(IAS)=NB
      IF (MODE.EQ.0) CALL INTERV (WSP(IAS),WSP(JBOUN),IBINTE,NINTE,WSP(I
     1AK),WSP(ITS))
C
C        THE NEW INDICES OF BUSES ARE CALCULATED
C
      CALL CHNUMB (NB,NINTE,NBOUND,NRET,IBINTE,WSP(JBOUN),WSP(JJBRET),NE
     1WNUM,IOLDNU)
C
C        RENUMBERING OF THE VECTORS LBINP,LBOUT
C
      CALL CHNGI (NTL,LBINP,NEWNUM)
      CALL CHNGI (NTL,LBOUT,NEWNUM)
C
      IF (IWRITE.NE.4) GO TO 30
      WRITE (OTPT,40)
      DO 20 I=1,NB
      WRITE (OTPT,50) I,NEWNUM(I)
   20 CONTINUE
   30 CONTINUE
      NEXT=NB-NINTE-NBOUND-NRET
      CALL SECOND (TTM2)
      TIM=TTM2-TTM1
      WRITE (OTPT,60) TIM
   40 FORMAT (//,3X,"INDICES OF BUSES ",//,2X,"OLD NUMB  NEW NUMB"/)
   50 FORMAT (2X,I5,5X,I5)
   60 FORMAT (//,2X,"CPU TIME FOR PARTITION OF THE SYSTEM ONTO THE INTER
     1NAL AND EXTERNAL SYSTEM =",F6.3,"SECONDS")
      RETURN
      END
C
C
      SUBROUTINE EQDRC1 (NBRE,NEXT,NEL,JRYT,ICYT,YT,NROW,NCOL,AK)
C
C        THIS SUBROUTINE EXTRACTS FROM A BUS ADMITTANCE MATRIX (VECTORS
C        YT,JRYT,ICYT) OF THE POWER SYSTEM A BUS ADMITTANCE MATRIX OF THE
C        THE INTERNAL SYSTEM (VECTORS YT,JRYT,ICYT) AND FORMS A
C        COEFFICIENT MATRIX (VECTORS AK,NROW,NCOL) DESCRIBING THE
C        EXTERNAL POWER SYSTEM
C
C        NBRE        NUMBER OF BOUNDARY AND RETAINED BUSES
C        NEXT        NUMBER OF BUSES OF THE EXTERNAL SYSTEM
C        NEL         NUMBER OF THE NONZERO ELEMENTS OF THE MATRIX AK
C
      INTEGER JRYT(1),ICYT(1),NROW(1),NCOL(1)
      COMPLEX Y,YT(1),AK(1)
C
      NEL=0
      L=NEXT+NBRE
      DO 20 I=1,NEXT
C
C        IN THIS LOOP THE "NEXT" ROWS OF THE ADMITTANCE MATRIX ARE
C        CHANGED TO THE FORM OF THE COEFFICIENT MATRIX
```

```
C
      J1=JRYT(I)
      J2=JRYT(I+1)-1
      DO 10 J=J1,J2
      KK=ICYT(J)
      NEL=NEL+1
      NROW(NEL)=I
      NCOL(NEL)=KK
      AK(NEL)=YT(J)
   10 CONTINUE
   20 CONTINUE
C
C        END OF THE CHANGES
C
      N=NEXT+1
      DO 40 I=N,L
C
C        IN THIS LOOP THE ROWS "N,...,L" OF THE ADMITTANCE MATRIX ARE
C        CHANGED TO THE FORM OF THE COEFFICIENT MATRIX
C
C        THE VALUES OF DIAGONAL ELEMENTS OF THE ADMITTANCE MATRIX OF THE
C        EXTERNAL SYSTEM CORRESPONDING TO THE BOUNDARY AND RETAINED BUSES
C        ARE CALCULATED
C
      J1=JRYT(I)+1
      J2=JRYT(I+1)-1
      J3=JRYT(I)
      Y=(0.0,0.0)
      DO 30 J=J1,J2
      KK=ICYT(J)
      IF (KK.GE.L) GO TO 30
      NEL=NEL+1
      NROW(NEL)=I
      NCOL(NEL)=KK
      AK(NEL)=YT(J)
      Y=Y+YT(J)
   30 CONTINUE
C
C        THE DIAGONAL ELEMENTS
C
      NEL=NEL+1
      NROW(NEL)=I
      NCOL(NEL)=I
      AK(NEL)=-Y
      YT(J3)=YT(J3)+Y
   40 CONTINUE
      RETURN
      END
C
C
      SUBROUTINE EQEXEQ (NB,NLB,NFIC,JRYT,ICYT,YT,BTYP,V,BCV,NBOUND,NRET
     1,NEXT,NEWNU,OLDNU,IBINTE,WSP,LSWSP,MODE,OTPT,IFLAG,IWRITE)
C
C        THIS SUBROUTINE FORMS THE BUS ADMITTACE MATRIX OF THE REDUCED
C        SYSTEM AND DETERMINES THE VECTOR OF BUS CONTROL VARIABLES
C        THE REDUCED SYSTEM CONSISTS OF THE INTERNAL SYSTEM AND THE FIXED
C        ADMITTANCE EQUIVALENT NETWORK CONNECTED TO THE INTERNAL SYSTEM
C        AT THE BOUNDARY BUSES
C        THE FIXED ADMITTANCE EQUIVALENT NETWORK CAN BE DETERMINED BY THE
C        SUBROUTINE AS A WARD EQUIVALENT
C
C        NB           NUMBER OF BUSES
C        NLB          NUMBER OF LOAD BUSES
C        NFIC         NUMBER OF FICTITIOUS BUSES
C        JRYT         VECTOR OF LENGTH NB+1. IT CONTAINS ROW INDICES OF
```

```
C                     A SPARSE BUS ADMITTANCE MATRIX
C         ICYT        VECTOR OF LENGTH JRYT(NB+1)-1. IT CONTAINS COLUMN
C                     INDICES OF A SPARSE ADMITTANCE MATRIX
C         YT          A SPARSE BUS ADMITTANCE MATRIX
C         BTYP        VECTOR OF BUS TYPES ( 0 LOAD BUS, 1 GENERATOR BUS,
C                     2 SLACK BUS )
C         V           COMPLEX BUS VOLTAGES (RECTANGULAR COORDINATES)
C         BCV         VECTOR OF LENGTH 2*NB. IT CONTAINS NOMINAL VALUES OF
C                     BUS CONTROL VARIABLES
C         NBOUND      NUMBER OF BOUNDARY BUSES
C         NRET        NUMBER OF RETAINED BUSES
C         NEWNU       VECTOR OF LENGTH NB. IT CONTAINS THE NEW INDICES OF
C                     BUSES
C         OLDNU       VECTOR OF LENGTH NB. IT CONTAINS THE OLD INDICES OF
C                     BUSES
C         IBINTE      VECTOR OF LENGTH NB. IT DETERMINES WHETHER A BUS
C                     BELONGS TO THE INTERNAL SYSTEM
C         WSP         REAL WORKSPACE
C         LWSP        LENGTH OF THE WORKSPACE WSP
C         MODE        TYPE OF AN EQUIVALENT :
C                     1 - STANDARD WARD EQUIVALENT WITHOUT SHUNTS
C                     2 - STANDARD WARD EQUIVALENT WITH BUFFER ZONE
C                     3 - EXTENDED WARD EQUIVALENT
C                     4 - SIMPLIFIED EXTENDED WARD EQUIVALENT
C                     5 - STANDARD WARD EQUIVALENT WITH SHUNTS IN THE
C                         EXTERNAL SYSTEM
C         OTPT        INDEX OF THE OUTPUT UNIT
C         IFLAG       RETURN FLAG :
C                     -1 LENGTH OF THE WORKSPACE FOR GAUSSIAN ELIMINATION
C                         IS TOO SMALL
C                     -3 LENGTH OF THE WORKSPACE IS TOO SMALL
C                      0 NORMAL RETURN
C         IWRITE      PARAMETER THAT CONTROLS OUTPUT
C
      INTEGER JRYT(1),ICYT(1),BTYP(1),IBINTE(1),OLDNU(1),NEWNU(1),OTPT
      REAL BCV(1),WSP(1)
      COMPLEX YT(1),V(1)
C
      CALL SECOND (TTM1)
      INB=NB
C
C       DISTRIBUTE THE WORKSPACE
C
      NBRE=NBOUND+NRET
      NEL=JRYT(NB+1)-1
      NFIC=0
      IFLAG=0
      JAP=1
      JNRB=JAP+4*NBRE
      IAD=JNRB+NB
      JNROW=IAD
      JNCOL=IAD+NEL
      JITAG=JNCOL+NEL
      JLCOL=JITAG+4*NEL
      JNON=JLCOL+4*NB
      JCX=JNON+NB
      JAK=JCX
      JDE=JAK+2*NEL
      JVR=JDE+4*NB
      JCE=JVR+2*NB
      IF (JCE+4*NEL.LE.LSWSP) GO TO 10
      IFLAG=-3
      RETURN
   10 CONTINUE
      NBCV=IAD
```

```
      N=4*NB
C
C        CHANGE THE ORDER IN THE MATRICES BTYP,BCV,V
C
      CALL CHNGJ (NB,BTYP,NEWNU,WSP(NBCV))
      CALL CHNGP (NB,BCV,NEWNU,WSP(NBCV))
      CALL CHNGC (NB,V,NEWNU,WSP(NBCV))
C
      DO 20 I=1,N
      WSP(JVR-1+I)=0.0
   20 CONTINUE
      IWSD=0
      IF (MODE.EQ.4)  IWSD=1
      IWREMS=1
      IF (MODE.EQ.5)  IWREMS=0
      IWESD=0
      IF (MODE.EQ.3.OR.MODE.EQ.4)  IWESD=1
C
C        COMPUTE THE COEFFICIENT MATRIX (VECTORS AK,NROW,NCOL)
C
      CALL CHNGAD (NB,NEL,JRYT,ICYT,YT,WSP(JNROW),WSP(JNCOL),WSP(JAK),NE
     1WNU)
C
      INEL=NEL
C
C        REMOVE ALL CONDUCTANCES FROM THE EXTERNAL SYSTEM: FOR THE
C        SIMPLIFIED WARD METHOD
C
      IF (IWSD.EQ.1) CALL EQWAES (NEL,NEXT,WSP(JNROW),WSP(JNCOL),WSP(JAK
     1))
C
C        COMPUTE A NEW ADMITTANCE MATRIX OF THE POWER SYSTEM
C
      CALL SPAMAT (NB,NEL,WSP(JNCOL),WSP(JNROW),WSP(JAK),JRYT,ICYT,YT)
C
      IF (IWESD.NE.1) GO TO 30
C
C        COMPUTE THE NUMBER AND VALUES OF FICTITIOUS BRANCHES
C
      CALL EQWEES (NB,NZ,NEXT,NBRE,NFIC,BTYP,JRYT,ICYT,YT,WSP(JNROW),WSP
     1(JNCOL),WSP(JAK),WSP(JVR),WSP(JLCOL),WSP(JITAG),WSP(JCE),WSP(JNRB)
     2,WSP(JAP),OTPT,IFLAG,IWRITE)
      IF (IFLAG.LT.0) RETURN
C
   30 CONTINUE
C
C        IF IWREMS=1 THE SHUNT ADMITTANCES ARE REMOVED
C
      IF (IWREMS.EQ.1) CALL EQREMS (NEXT,JRYT,ICYT,YT)
C
C        COMPUTE THE COEFFICIENT MATRIX (VECTORS WSP(AK),WSP(NROW) AND
C        WSP(NCOL)) FOR THE EXTERNAL POWER SYSTEM AND THE ADMITTANCE
C        MATRIX (VECTORS YT,JRYT,ICYT) OF THE INTERNAL POWER SYSTEM
C
      CALL EQDRC1 (NBRE,NEXT,NEL,JRYT,ICYT,YT,WSP(JNROW),WSP(JNCOL),WSP(
     1JAK))
C
      IF (IWSD.EQ.1) CALL EQWAES (NEL,NEXT,WSP(JNROW),WSP(JNCOL),WSP(JAK
     1))
      NB1=NBRE+NEXT
C
C        PERFORM GAUSSIAN ELIMINATION
C
      CALL GAUSEL (NB1,NEL,NEXT,WSP(JNROW),WSP(JNCOL),WSP(JAK),WSP(JVR),
     1WSP(JLCOL),WSP(JITAG),WSP(JCE),IFLAG,IWRITE)
```

```
      IF (IFLAG.LT.0) RETURN
C
C        ATTACH THE EXTERNAL EQUIVALENT TO THE INTERNAL POWER SYSTEM
C        AT BOUNDARY BUSES
C
      CALL EQINRC (NB,NEXT,NBRE,NEL,JRYT,ICYT,YT,WSP(JNROW),WSP(JNCOL),W
     1SP(JAK))
      N=NB-NEXT
C
C        ATTACH THE FICTITIOUS BRANCHES TO THE BOUNDARY BUSES
C        (FOR THE EXTENDED AND SIMPLIFIED WARD METHOD)
C
      IF (IWESD.EQ.1) CALL EQINBR (N,NEL,NFIC,WSP(JNROW),WSP(JNCOL),WSP(
     1JNRB),WSP(JAK),WSP(JAP))
C
C        CHANGE THE COEFFICIENT MATRIX (VECTORS WSP(JAK),WSP(JNROW) AND
C        WSP(JNCOL)) TO THE FORM OF THE SPARSE MATRIX (VECTORS YT,JRYT,
C        ICYT)
C
      CALL SPAMAT (N,NEL,WSP(JNCOL),WSP(JNROW),WSP(JAK),JRYT,ICYT,YT)
C
      CALL PRINAD (N,JRYT,ICYT,YT,IWRITE,OTPT)
C
C        COMPUTE THE NEW VALUES OF THE BUS CONTROL VARIABLES OF THE
C        REDUCED POWER SYSTEM
C
      CALL EQBCV (NB,NLB,NEXT,NFIC,BTYP,BCV,V,WSP(NBCV),YT,JRYT,ICYT,OLD
     1NU,IBINTE,WSP(JNRB),OTPT,IWRITE)
C
      CALL SECOND (TTM2)
      TIEL=TTM2-TTM1
      WRITE (OTPT,40) TIEL
   40 FORMAT (//,2X,"CPU TIME FOR POWER NETWORK EQUIVALENT = ",F6.3,"SEC
     1ONDS"/)
      RETURN
      END
C
C
      SUBROUTINE EQFORM (NB,NBEQ,NLBE,NEXT,NFIC,NBOUND,JBOUND,NRET,JRET,
     1BTYP,BCV,V,NEWNUM,OLDNUM,JBINTE,YT,JRYT,ICYT,LBINP,LBOUT,NTL,MODE,
     2OTPT,IWRITE)
C
C        THE SUBROUTINE EQFORM CREATES A FORMATTED DATA FILE DESCRIBING
C        THE REDUCED POWER SYSTEM. THE DATA IS PRINTED OUT ON THE UNIT
C        OTPT
C
C        NB          NUMBER OF BUSES OF THE UNREDUCED POWER SYSTEM
C        NBEQ        NUMBER OF BUSES OF THE REDUCED POWER SYSTEM
C        NLBE        NUMBER OF LOAD BUSES OF THE REDUCED POWER SYSTEM
C        NEXT        NUMBER OF BUSES OF THE EXTERNAL POWER SYSTEM
C        NFIC        NUMBER OF FICTITIOUS BUSES
C        NBOUND      NUMBER OF BOUNDARY BUSES
C        JBOUND      VECTOR OF LENGTH NBOUND. IT CONTAINS BOUNDARY BUSES
C        NRET        NUMBER OF RETAINED BUSES
C        JRET        VECTOR OF LENGTH NRET. IT CONTAINS RETAINED BUSES
C        BTYP        VECTOR OF BUS TYPES ( 0 LOAD BUS, 1 GENERATOR BUS,
C                    2 SLACK BUS ) OF THE REDUCED POWER SYSTEM
C        BCV         VECTOR OF LENGTH 2*NB. IT CONTAINS NOMINAL VALUES OF
C                    BUS CONTROL VARIABLES OF THE REDUCED POWER SYSTEM
C        V           COMPLEX BUS VOLTAGES (RECTANGULAR COORDINATES) OF THE
C                    REDUCED POWER SYSTEM
C        NEWNUM      VECTOR OF LENGTH NB. IT STORES THE NEW INDICES OF
C                    BUSES
C        IOLDNU      VECTOR OF LENGTH NB. IT STORES THE OLD INDICES OF
C                    BUSES W.R.T. THE NEW INDICES
```

```
C       IBINTE      VECTOR OF LENGTH NB. IT STORES BUSES OF THE INTERNAL
C                   POWER SYSTEM
C       YT          A SPARSE BUS ADMITTANCE MATRIX
C       JRYT        VECTOR OF LENGTH NB+1. IT CONTAINS ROW INDICES OF
C                   THE SPARSE BUS ADMITTANCE MATRIX
C       ICYT        VECTOR OF LENGTH JRYT(NB+1)-1. IT CONTAINS COLUMN
C                   INDICES OF THE SPARSE ADMITTANCE MATRIX
C       LBINP,LBOUT     VECTORS OF LENGTH NTL. LBINP(K),LBOUT(K) CONTAIN
C                       INDICES OF BUSES INCIDENT WITH THE KTH LINE
C       NTL         NUMBER OF TRANSMISSION LINES
C       MODE        TYPE OF AN EQUIVALENT :
C                   1 - STANDARD WARD EQUIVALENT WITHOUT SHUNTS
C                   2 - STANDARD WARD EQUIVALENT WITH BUFFER ZONE
C                   3 - EXTENDED WARD EQUIVALENT
C                   4 - SIMPLIFIED EXTENDED WARD EQUIVALENT
C                   5 - STANDARD WARD EQUIVALENT WITH SHUNTS IN THE
C                       EXTERNAL SYSTEM
C       OTPT        INDEX OF THE OUTPUT UNIT
C       IWRITE      PARAMETER THAT CONTROLS OUTPUT
C                   1 - DESCRIPTION OF THE REDUCED POWER SYSTEM IS
C                       PRINTED OUT ON THE UNIT OTPT
C
      INTEGER BTYP(1),JBOUND(1),JRET(1),NEWNUM(1),OLDNUM(1),JRYT(1),ICYT
     1(1),JBINTE(1),LBINP(1),LBOUT(1),OTPT,PAR(8)
      REAL BCV(1),TYPM(6),DESC(8),PL(16)
      COMPLEX V(1),YT(1)
C
      DATA TYPM/"SWARD1","BZWARD","EXWARD","SIWARD","SWARD2","FDWARD"/,D
     1ESC/"     NB= ","  NBEQ= ","  NLBE= ","  NEXT= ","  NFIC= ","NBOUND
     2= ","  NRET= ","   NTL= "/,PL/"LBINP ","LBOUT ","NEWNUM","OLDNUM",
     3"JBINTE","BTYP","REALV","IMAGV","BCV1 ","BCV2 ","JRYT  ","  ICYT",
     4"REAL YT","IMAG YT","BOUNDARY","RETAINED"/
C
      IF (IWRITE.NE.1) RETURN
C
C       THE PARAMETERS DESCRIBING THE REDUCED POWER SYSTEM ARE PRINTED
C       OUT ON THE UNIT OTPT
C
      WRITE (OTPT,130) TYPM(MODE),MODE
      PAR(1)=NB
      PAR(2)=NBEQ
      PAR(3)=NLBE
      PAR(4)=NEXT
      PAR(5)=NFIC
      PAR(6)=NBOUND
      PAR(7)=NRET
      PAR(8)=NTL
      DO 10 I=1,8
      WRITE (OTPT,150) DESC(I),PAR(I)
   10 CONTINUE
C
      IF (NBOUND.EQ.0) GO TO 20
      WRITE (OTPT,160) PL(15)
      WRITE (OTPT,80) (JBOUND(I),I=1,NBOUND)
   20 CONTINUE
C
      IF (NRET.EQ.0) GO TO 30
      WRITE (OTPT,160) PL(16)
      WRITE (OTPT,80) (JRET(I),I=1,NRET)
   30 CONTINUE
C
      WRITE (OTPT,170) (PL(I),I=3,5)
C
      DO 40 I=1,NB
      WRITE (OTPT,100) NEWNUM(I),OLDNUM(I),JBINTE(I)
```

```
   40 CONTINUE
      WRITE (OTPT,180) (PL(I),I=6,10)
      DO 50 I=1,NBEQ
      WRITE (OTPT,110) BTYP(I),V(I),BCV(2*I-1),BCV(2*I)
   50 CONTINUE
      WRITE (OTPT,170) PL(11)
      K=NBEQ+1
      WRITE (OTPT,80) (JRYT(I),I=1,K)
      K=JRYT(K)-1
      WRITE (OTPT,190) (PL(I),I=12,14)
      DO 60 I=1,K
      WRITE (OTPT,120) ICYT(I),YT(I)
   60 CONTINUE
      WRITE (OTPT,140) PL(1),PL(2)
      DO 70 I=1,NTL
      WRITE (OTPT,90) LBINP(I),LBOUT(I)
   70 CONTINUE
C
C       END OF PRINTING
C
      RETURN
   80 FORMAT (1X,10I5)
   90 FORMAT (1X,2I5)
  100 FORMAT (1X,3(1X,I5,1X))
  110 FORMAT (1X,I2,2X,4(E12.6,4X))
  120 FORMAT (1X,I5,2X,2(E12.6,2X))
  130 FORMAT (1X,A6,3X,I2)
  140 FORMAT (1X,2A6)
  150 FORMAT (1X,A8,I4)
  160 FORMAT (1X,A8)
  170 FORMAT (1X,3(A6,1X))
  180 FORMAT (1X,A4,4(5X,A5,6X))
  190 FORMAT (1X,A6,7X,A7,6X,A7)
      END
C
C
      SUBROUTINE EQINBR (NB,NEL,NFIC,NROW,NCOL,NRB,AK,AP)
C
C       THIS SUBROUTINE ATTACHES FICTITIOUS BRANCHES TO THE
C       RELEVANT BOUNDARY BUSES
C
C       NB          NUMBER OF BUSES
C       NFIC        NUMBER OF FICTITIOUS BUSES
C       NEL         NUMBER OF NONZERO ELEMENTS OF THE COEFFICIENT MATRIX
C       AK          VECTOR OF LENGTH NEL. IT CONTAINS THE NONZERO
C                   ELEMENTS OF THE COEFFICIENT MATRIX
C       NROW,NCOL   VECTORS OF LENGTH NEL. THEY CONTAIN ROW AND COLUMN
C                   INDICES OF THE NONZERO ELEMENTS OF THE COEFFICIENT
C                   MATRIX
C       NRB         VECTOR OF LENGTH NFIC. IT CONTAINS THE INDICES OF
C                   BUSES INCIDENT WITH THE FICTITIOUS BRANCHES
C       AP          VECTOR OF LENGTH NFIC. IT CONTAINS THE VALUES OF THE
C                   FICTITIOUS BRANCHES
C
      INTEGER NROW(1),NCOL(1),NRB(1)
      COMPLEX AK(1),AP(1)
C
C       CHANGE THE ROW INDICES AND THE COLUMN INDICES OF THE COEFFICIENT
C       MATRIX AK
C
      DO 10 I=1,NEL
      IF (NROW(I).EQ.NB) NROW(I)=NROW(I)+NFIC
      IF (NCOL(I).EQ.NB) NCOL(I)=NCOL(I)+NFIC
   10 CONTINUE
C
```

```
C        COMPUTE THE NEW ELEMENTS OF THE COEFFICIENT MATRIX AK AND
C        THE MATRICES NROW AND NCOL
C
      DO 20 I=1,NFIC
C
C        IN THIS LOOP THE NEW DIAGONAL ELEMENTS OF THE COEFFICIENT MATRIX
C        ARE CALCULATED
C
      NEL=NEL+1
      NROW(NEL)=NB+I-1
      NCOL(NEL)=NB+I-1
      AK(NEL)=CMPLX(0.0,AIMAG(AP(I)))
   20 CONTINUE
C
      DO 50 I=1,NFIC
C
C        IN THIS LOOP THE NEW ELEMENTS OF THE COEFFICIENT MATRIX
C        ARE CALCULATED AND THE DIAGONAL ELEMENTS ARE MODIFIED
C
      NEL=NEL+1
      NROW(NEL)=NRB(I)
      NCOL(NEL)=NB+I-1
      AK(NEL)=-CMPLX(0.0,AIMAG(AP(I)))
      KK=NRB(I)
      DO 30 K=1,NEL
      IF (NROW(K).NE.KK) GO TO 30
      IF (NCOL(K).NE.KK) GO TO 30
      AK(K)=AK(K)+CMPLX(0.0,AIMAG(AP(I)))
      GO TO 40
   30 CONTINUE
   40 CONTINUE
      NEL=NEL+1
      NROW(NEL)=NB+I-1
      NCOL(NEL)=NRB(I)
      AK(NEL)=-CMPLX(0.0,AIMAG(AP(I)))
   50 CONTINUE
      NB=NB+NFIC
      RETURN
      END
C
C
      SUBROUTINE EQMOSO (INB,NB,NFIC,NEXT,V,AXV,JOLDNU,MODE)
C
C        THIS SUBROUTINE PREPARES DATA FOR SUBROUTINE EQCONTI.
C        IT CHANGES THE ORDER OF INB ELEMENTS STORED IN THE VECTOR V
C        ACCORDING TO THE OLD INDICES STORED BY THE VECTOR JOLDNU
C
C        INB          NUMBER OF BUSES OF THE UNREDUCED POWER SYSTEM
C        NB           NUMBER OF BUSES OF THE REDUCED POWER SYSTEM
C        NEXT         NUMBER OF BUSES OF THE EXTERNAL POWER SYSTEM
C        NFIC         NUMBER OF FICTITIOUS BUSES
C        V            COMPLEX BUS VOLTAGES (RECTANGULAR COORDINATES) OF THE
C                     REDUCED POWER SYSTEM
C        JOLDNU       VECTOR OF LENGTH INB. IT CONTAINS THE OLD INDICES OF
C                     BUSES W.R.T. THE NEW INDICES
C        AXV          AN AUXILIARY VECTOR
C        MODE         TYPE OF AN EQUIVALENT
C
      INTEGER JOLDNU(1)
      COMPLEX V(1)
      REAL AXV(1)
C
      K=2*INB
      DO 10 I=1,K
      AXV(I)=0.0
```

```
   10 CONTINUE
      NBF=NB-NFIC-1
      DO 20 I=1,NBF
      J=NEXT+I
      IF (MODE.NE.0) K=JOLDNU(J)
      AXV(2*K-1)=REAL(V(I))
      AXV(2*K)=AIMAG(V(I))
   20 CONTINUE
      AXV(2*INB-1)=REAL(V(NB))
      AXV(2*INB)=AIMAG(V(NB))
      DO 30 I=1,INB
      V(I)=CMPLX(AXV(2*I-1),AXV(2*I))
   30 CONTINUE
      RETURN
      END
C
C
      SUBROUTINE EQINRC (NB,NEXT,NBRE,NEL,JRYT,ICYT,YT,NROW,NCOL,AK)
C
C         THIS SUBROUTINE LINKS AN EXTERNAL EQUIVALENT DESCRIBED BY A
C         COEFFICIENT MATRIX (VECTORS: AK,NROW,NCOL) WITH THE INTERNAL
C         SYSTEM DESCRIBED BY AN ADMITTANCE MATRIX STORED IN THE
C         SPARSE FORM (VECTORS YT,JRYT,ICYT). THE REDUCED SYSTEM IS
C         STORED IN THE COEFFICIENT MATRIX (VECTORS AK,NROW,NCOL)
C
      INTEGER JRYT(1),ICYT(1),NROW(1),NCOL(1)
      COMPLEX YT(1),AK(1)
C
C         MODIFICATION OF THE MATRIX AK AND THE MATRICES NROW AND NCOL
C
      N=NEXT+1
      NELK=NEL
      L=NEXT+NBRE
      DO 50 I=N,NB
      J1=JRYT(I)
      J2=JRYT(I+1)-1
      K1=I-NEXT
      DO 40 J=J1,J2
      KK=ICYT(J)
      K2=KK-NEXT
      IF (KK.GT.L) GO TO 20
      IF (KK.LT.N) GO TO 40
      DO 10 I1=1,NEL
      IF (K1.EQ.NROW(I1).AND.K2.EQ.NCOL(I1)) GO TO 30
   10 CONTINUE
   20 CONTINUE
      NELK=NELK+1
      NROW(NELK)=K1
      NCOL(NELK)=K2
      AK(NELK)=YT(J)
      GO TO 40
   30 CONTINUE
      AK(I1)=AK(I1)+YT(J)
   40 CONTINUE
   50 CONTINUE
      NEL=NELK
      RETURN
      END
C
C
      SUBROUTINE EQREAD (NB,NBEQ,NLBE,NEXT,NFIC,NBOUND,JBOUND,NRET,JRET,
     1BTYP,BCV,V,NEWNUM,OLDNUM,JBINTE,YT,JRYT,ICYT,LBINP,LBOUT,NTL,MODE,
     2INPT,OTPT,IWRITE)
C
C       SUBROUTINE EQREAD READS INPUT DATA DESCRIBING THE REDUCED POWER
```

```
C        SYSTEM FROM A FILE WHICH IS CREATED BY THE SUBROUTINE EQFORM
C
C        NB            NUMBER OF BUSES OF THE UNREDUCED POWER SYSTEM
C        NBEQ          NUMBER OF BUSES OF THE REDUCED POWER SYSTEM
C        NLBE          NUMBER OF LOAD BUSES OF THE REDUCED POWER SYSTEM
C        NEXT          NUMBER OF BUSES OF THE EXTERNAL POWER SYSTEM
C        NFIC          NUMBER OF FICTITIOUS BUSES
C        NBOUND        NUMBER OF BOUNDARY BUSES
C        JBOUND        VECTOR OF LENGTH NBOUND. IT CONTAINS BOUNDARY BUSES
C        NRET          NUMBER OF RETAINED BUSES
C        JRET          VECTOR OF LENGTH NRET. IT CONTAINS RETAINED BUSES
C        BTYP          VECTOR OF BUS TYPES ( 0 LOAD BUS, 1 GENERATOR BUS,
C                      2 SLACK BUS ) OF THE REDUCED POWER SYSTEM
C        BCV           VECTOR OF LENGTH 2*NB. IT CONTAINS NOMINAL VALUES
C                      OF BUS CONTROL VARIABLES OF THE REDUCED POWER
C                      SYSTEM
C        V             COMPLEX BUS VOLTAGES (RECTANGULAR COORDINATES) OF THE
C                      REDUCED POWER SYSTEM
C        NEWNUM        VECTOR OF LENGTH NB. IT STORES THE NEW INDICES OF
C                      BUSES
C        IOLDNU        VECTOR OF LENGTH NB. IT STORES THE OLD INDICES OF
C                      BUSES W.R.T. THE NEW INDICES
C        IBINTE        VECTOR OF LENGTH NB. IT STORES BUSES OF THE INTERNAL
C                      POWER SYSTEM
C        YT            A SPARSE BUS ADMITTANCE MATRIX
C        JRYT          VECTOR OF LENGTH NB+1. IT CONTAINS ROW INDICES OF
C                      THE SPARSE BUS ADMITTANCE MATRIX
C        ICYT          VECTOR OF LENGTH JRYT(NB+1)-1. IT CONTAINS COLUMN
C                      INDICES OF THE SPARSE ADMITTANCE MATRIX
C        LBINP,LBOUT       VECTORS OF LENGTH NTL. LBINP(K),LBOUT(K) CONTAIN
C                          INDICES OF BUSES INCIDENT WITH THE KTH LINE
C        NTL           NUMBER OF TRANSMISSION LINES
C        MODE          TYPE OF AN EQUIVALENT :
C                      1 - STANDARD WARD EQUIVALENT WITHOUT SHUNTS
C                      2 - STANDARD WARD EQUIVALENT WITH BUFFER ZONE
C                      3 - EXTENDED WARD EQUIVALENT
C                      4 - SIMPLIFIED EXTENDED WARD EQUIVALENT
C                      5 - STANDARD WARD EQUIVALENT WITH SHUNTS IN THE
C                          EXTERNAL  SYSTEM
C        INPT          INDEX OF THE INPUT UNIT
C        OTPT          INDEX OF THE OUTPUT UNIT
C        IWRITE        PARAMETER THAT CONTROLS OUTPUT
C                      1 - DESCRIPTION OF THE REDUCED POWER SYSTEM IS
C                          PRINTED OUT ON THE UNIT OTPT
C
      INTEGER BTYP(1),JBOUND(1),JRET(1),NEWNUM(1),OLDNUM(1),JRYT(1),ICYT
     1(1),JBINTE(1),LBINP(1),LBOUT(1),INPT,PAR(8),OTPT
      REAL BCV(1),DESC(8),PL(16)
      COMPLEX V(1),YT(1)
C
C        THE DATA DESCRIBING THE REDUCED POWER SYSTEM IS READ FROM THE
C        UNIT INPT
C
      READ (INPT,170) RTYP,K
      MODE=K
      IF (IWRITE.EQ.1) WRITE (OTPT,170) RTYP,MODE
      DO 10 I=1,8
      READ (INPT,190) DESC(I),PAR(I)
   10 CONTINUE
      IF (IWRITE.NE.1) GO TO 30
      DO 20 I=1,8
      WRITE (OTPT,190) DESC(I),PAR(I)
   20 CONTINUE
   30 CONTINUE
      NB=PAR(1)
```

```
      NBEQ=PAR(2)
      NLBE=PAR(3)
      NEXT=PAR(4)
      NFIC=PAR(5)
      NBOUND=PAR(6)
      NRET=PAR(7)
      NTL=PAR(8)
C
      IF (NBOUND.EQ.0) GO TO 40
      READ (INPT,200) PL(15)
      IF (IWRITE.EQ.1) WRITE (OTPT,200) PL(15)
      READ (INPT,120) (JBOUND(I),I=1,NBOUND)
      IF (IWRITE.EQ.1) WRITE (OTPT,120) (JBOUND(I),I=1,NBOUND)
   40 CONTINUE
C
      IF (NRET.EQ.0) GO TO 50
      READ (INPT,200) PL(16)
      IF (IWRITE.EQ.1) WRITE (OTPT,200) PL(16)
      READ (INPT,120) (JRET(I),I=1,NRET)
      IF (IWRITE.EQ.1) WRITE (OTPT,120) (JRET(I),I=1,NRET)
   50 CONTINUE
C
      READ (INPT,210) (PL(I),I=3,5)
C
      IF (IWRITE.EQ.1) WRITE (OTPT,210) (PL(I),I=3,5)
      DO 60 I=1,NB
      READ (INPT,140) NEWNUM(I),OLDNUM(I),JBINTE(I)
   60 CONTINUE
      IF (IWRITE.NE.1) GO TO 80
      DO 70 I=1,NB
      WRITE (OTPT,140) NEWNUM(I),OLDNUM(I),JBINTE(I)
   70 CONTINUE
   80 CONTINUE
C
      READ (INPT,220) (PL(I),I=6,10)
      IF (IWRITE.EQ.1) WRITE (OTPT,220) (PL(I),I=6,10)
C
      DO 90 I=1,NBEQ
      READ (INPT,150) BTYP(I),V(I),BCV(2*I-1),BCV(2*I)
      IF (IWRITE.EQ.1) WRITE (OTPT,150) BTYP(I),V(I),BCV(2*I-1),BCV(2*I)
   90 CONTINUE
C
      READ (INPT,210) PL(11)
      IF (IWRITE.EQ.1) WRITE (OTPT,210) PL(11)
      K=NBEQ+1
C
      READ (INPT,120) (JRYT(I),I=1,K)
      IF (IWRITE.EQ.1) WRITE (OTPT,120) (JRYT(I),I=1,K)
      K=JRYT(K)-1
C
      READ (INPT,230) (PL(I),I=12,14)
      IF (IWRITE.EQ.1) WRITE (OTPT,230) (PL(I),I=12,14)
      DO 100 I=1,K
C
      READ (INPT,160) ICYT(I),YT(I)
      IF (IWRITE.EQ.1) WRITE (OTPT,160) ICYT(I),YT(I)
  100 CONTINUE
C
      READ (INPT,180) PL(1),PL(2)
      IF (IWRITE.EQ.1) WRITE (OTPT,180) PL(1),PL(2)
C
      DO 110 I=1,NTL
      READ (INPT,130) LBINP(I),LBOUT(I)
      IF (IWRITE.EQ.1) WRITE (OTPT,130) LBINP(I),LBOUT(I)
  110 CONTINUE
```

```
C
C        END OF READING
C
  120 FORMAT (1X,10I5)
  130 FORMAT (1X,2I5)
  140 FORMAT (1X,I6,I7,I7)
  150 FORMAT (1X,I2,2X,4(E12.6,4X))
  160 FORMAT (1X,I5,2X,2(E12.6,2X))
  170 FORMAT (1X,A6,3X,I2)
  180 FORMAT (1X,2A6)
  190 FORMAT (1X,A8,I4)
  200 FORMAT (1X,A8)
  210 FORMAT (1X,3(A6,1X))
  220 FORMAT (1X,A4,4(5X,A5,6X))
  230 FORMAT (1X,A6,7X,A7,6X,A7)
      RETURN
      END
```

```
C
C
      SUBROUTINE EQREEY (I1,JY1,NB,NEL,JRYT,ICYT,YT)
C
C       THIS SUBROUTINE REMOVES AN ELEMENT JY1 FROM VECTORS ICYT ,YT AND
C       MODIFIES THE VECTOR JRYT
C
      INTEGER JRYT(1),ICYT(1)
      COMPLEX YT(1)
      NEL=NEL-1
      DO 10 I=JY1,NEL
      ICYT(I)=ICYT(I+1)
      YT(I)=YT(I+1)
   10 CONTINUE
      K=I1+1
      N=NB+1
      DO 20 I=K,N
      JRYT(I)=JRYT(I)-1
   20 CONTINUE
      RETURN
      END
C
C
      SUBROUTINE EQREMS (NEXT,JRYT,ICYT,YT)
C
C       THIS SUBROUTINE CHANGES THE VALUE OF THE "NEXT" DIAGONAL
C       ELEMENTS OF THE ADMITTANCE MATRIX YT STORED IN THE SPARSE FORM
C       (VECTORS YT,JRYT,ICYT)
C
      INTEGER JRYT(1),ICYT(1)
      COMPLEX YT(1),Y
C
      DO 20 I=1,NEXT
      J1=JRYT(I)+1
      J2=JRYT(I+1)-1
      Y=(0.0,0.0)
C
C       COMPUTE THE SUM OF THE ELEMENTS OF THE ITH ROW OF THE MATRIX YT
C       WITHOUT THE DIAGONAL ELEMENT
C
      DO 10 J=J1,J2
      Y=Y+YT(J)
   10 CONTINUE
      J=JRYT(I)
C
C       CHANGE THE ITH DIAGONAL ELEMENT OF THE MATRIX YT
C
      YT(J)=-Y
   20 CONTINUE
      RETURN
      END
C
C
      SUBROUTINE EQWAES (NEL,NEXT,NROW,NCOL,AK)
C
C       THIS SUBROUTINE SETS TO ZERO THE REAL PART OF THE COEFFICIENTS
C       OF THE MATRIX AK WHOSE EITHER ROW OR COLUMN INDICES ARE LESSER
C       THAN OR EQUAL TO THE "NEXT" PARAMETER
C
C       NEL        NUMBER OF THE NONZERO ELEMENTS OF THE MATRIX AK
C       NEXT       NUMBER OF THE EXTERNAL BUSES
C
      INTEGER NROW(1),NCOL(1)
      COMPLEX AK(1)
C
```

```
      DO 50 I=1,NEL
      IF (NROW(I).LE.NEXT.AND.NCOL(I).LE.NEXT) GO TO 40
      IF (NROW(I).LE.NEXT.AND.NCOL(I).GT.NEXT) GO TO 40
      IF (NCOL(I).LE.NEXT.AND.NROW(I).GT.NEXT) GO TO 10
      GO TO 50
   10 CONTINUE
C
C        FIND DIAGONAL ELEMENT
C
      K=NROW(I)
      DO 20 J=1,NEL
      IF (NROW(J).EQ.K.AND.NCOL(J).EQ.K) GO TO 30
   20 CONTINUE
      GO TO 50
   30 CONTINUE
      R1=REAL(AK(I))
      R=REAL(AK(J))
      AI=AIMAG(AK(J))
      R=R1+R
C
C        CHANGE THE REAL PART OF THE COEFFICIENT AK(J)
C
      AK(J)=CMPLX(R,AI)
      AK(I)=CMPLX(0.0,AIMAG(AK(I)))
      GO TO 50
   40 CONTINUE
      X=AIMAG(AK(I))
C
C        SET TO ZERO THE REAL PART OF THE COEFFICIENT AK(I)
C
      AK(I)=CMPLX(0.0,X)
   50 CONTINUE
      RETURN
      END
C
C
      SUBROUTINE EQWARD (NB,NLB,NTL,LBINP,LBOUT,YT,JRYT,ICYT,BTYP,V,BCV,
     1NBOUND,JBOUND,NRET,JRET,NEWNUM,IOLDNU,IBINTE,NEXT,NFIC,WSP,LWSP,MO
     2DE1,MODE2,MODE3,OTPT,INPT,IFLAG,IWRITE)
C
C        THIS IS THE MAIN SUBROUTINE OF THE WARDEQ PACKAGE. IT CALLS
C        SUBROUTINES EQDEIE AND EQEXEQ,CHECKS THE CORRECTNESS OF THE
C        ARGUMENTS AND SUPPLIES THE VALUE OF IFLAG
C
C
C        NB            NUMBER OF BUSES
C        NLB           NUMBER OF LOAD BUSES
C        NTL           NUMBER OF TRANSMISSION LINES
C        LBINP,LBOUT       VECTORS OF LENGTH NTL. LBINP(K),LBOUT(K) CONTAIN
C                          INDICES OF BUSES INCIDENT WITH THE KTH LINE
C        JRYT          VECTOR OF LENGTH NB+1. IT CONTAINS ROW INDICES OF
C                      A SPARSE BUS ADMITTANCE MATRIX
C        ICYT          VECTOR OF LENGTH JRYT(NB+1)-1. IT CONTAINS COLUMN
C                      INDICES OF A SPARSE ADMITTANCE MATRIX
C        YT            A SPARSE BUS ADMITTANCE MATRIX
C        BTYP          VECTOR OF BUS TYPES ( 0 LOAD BUS, 1 GENERATOR BUS,
C                      2 SLACK BUS )
C        V             COMPLEX BUS VOLTAGES (RECTANGULAR COORDINATES)
C        BCV           VECTOR OF LENGTH 2*NB. IT CONTAINS NOMINAL VALUES
C                      OF BUS CONTROL VARIABLES
C        NBOUND        NUMBER OF BOUNDARY BUSES
C        JBOUND        VECTOR OF LENGTH NBOUND. IT CONTAINS BOUNDARY BUSES
C        NRET          NUMBER OF RETAINED BUSES
C        JRET          VECTOR OF LENGTH NRET. IT CONTAINS RETAINED BUSES
C        NEWNUM        VECTOR OF LENGTH NB. IT STORES THE NEW INDICES OF
```

```
C                   BUSES
C       IOLDNU      VECTOR OF LENGTH NB. IT STORES THE OLD INDICES OF
C                   BUSES W.R.T. THE NEW INDICES
C       IBINTE      VECTOR OF LENGTH NB. IBINTE(K) IS SET TO 1 IF THE KTH
C                   BUS BELONGS TO THE INTERNAL POWER SYSTEM
C       NEXT        NUMBER OF BUSES BELONGING TO THE EXTERNAL POWER
C                   SYSTEM
C       NFIC        NUMBER OF FICTITIOUS BUSES
C       WSP         REAL WORKSPACE
C       LWSP        LENGTH OF THE WORKSPACE WSP
C       MODE1       MODE OF AN OPERATION
C                   0 - NORMAL ENTRY
C                   1 - ENTRY ONLY TO THE SUBROUTINE EQEXEQ
C       MODE2       MODE OF AN OPERATION
C                   0 - A REDUCED POWER SYSTEM IS DETERMINED
C                   1 - AN EXTERNAL EQUIVALENT IS DETERMINED
C       MODE3       TYPE OF AN EQUIVALENT :
C                   1 - STANDARD WARD EQUIVALENT WITHOUT SHUNTS
C                   2 - STANDARD WARD EQUIVALENT WITH BUFFER ZONE
C                   3 - EXTENDED WARD EQUIVALENT
C                   4 - SIMPLIFIED EXTENDED WARD EQUIVALENT
C                   5 - STANDARD WARD EQUIVALENT WITH SHUNTS IN THE
C                       EXTERNAL SYSTEM
C       OTPT        INDEX OF THE OUTPUT UNIT
C       INPT        INDEX OF THE INPUT UNIT
C       IFLAG       RETURN FLAG:
C                    0 - NORMAL RETURN
C                   -1 - THE LENGTH OF THE WORKSPACE FOR GAUSSIAN
C                        ELIMINATION IS TOO SMALL
C                   -2 - THE LENGTH OF THE WORKSPACE IS TOO SMALL
C                   -3 - THE LENGTH OF THE WORKSPACE FOR THE SUBROUTINE
C                        EQEXEQ IS TOO SMALL
C                   -5 - THE PARAMETER MODE1 IS INCORRECT
C                   -6 - THE PARAMETER MODE2 IS INCORRECT
C                   -7 - THE PARAMETER MODE3 IS INCORRECT
C       IWRITE      PARAMETER THAT CONTROLS OUTPUT
C
      INTEGER LBINP(1),LBOUT(1),JRYT(1),ICYT(1),BTYP(1),JBOUND(1),JRET(1
     1),NEWNUM(1),IOLDNU(1),IBINTE(1),OTPT
      REAL BCV(1),WSP(1)
      COMPLEX V(1),YT(1)
C
      IFLAG=0
C
C       CHECK THE MODE PARAMETERS
C
      IF (MODE1.EQ.1) GO TO 80
      IF (MODE1.GT.1.OR.MODE1.LT.0) GO TO 90
      IF (MODE2.GT.1.OR.MODE2.LT.0) GO TO 100
      IF (MODE3.GT.5.OR.MODE3.LE.0) GO TO 110
      IF (MODE2.EQ.0) WRITE (OTPT,190)
      IF (MODE2.EQ.1) WRITE (OTPT,200)
      GO TO (10,20,30,40,50), MODE3
   10 WRITE (OTPT,140) NB
      GO TO 60
   20 WRITE (OTPT,150) NB
      GO TO 60
   30 WRITE (OTPT,160) NB
      GO TO 60
   40 WRITE (OTPT,170) NB
      GO TO 60
   50 CONTINUE
      WRITE (OTPT,180) NB
   60 CONTINUE
      IF (NBOUND.EQ.0) GO TO 70
```

```
      WRITE (OTPT,120) (JBOUND(I),I=1,NBOUND)
      IF (NRET.EQ.0) GO TO 70
      WRITE (OTPT,130) (JRET(I),I=1,NRET)
   70 CONTINUE
C
C        PARTITIONING OF THE UNREDUCED POWER SYSTEM AND PREPARING
C        THE RENUMBERING OF THE BUSES IN THE SYSTEM
C
      INB=NB
      CALL EQDEIE (NB,NTL,LBINP,LBOUT,NBOUND,JBOUND,NRET,JRET,NEWNUM,IOL
     1DNU,IBINTE,NEXT,WSP,LWSP,MODE2,OTPT,IFLAG,IWRITE)
      IF (IFLAG.LT.0) RETURN
   80 CONTINUE
C
C        THE LOAD FLOW PROBLEM FOR THE REDUCED SYSTEM IS FORMULATED
C
      CALL EQEXEQ (NB,NLB,NFIC,JRYT,ICYT,YT,BTYP,V,BCV,NBOUND,NRET,NEXT,
     1NEWNUM,IOLDNU,IBINTE,WSP,LWSP,MODE3,OTPT,IFLAG,IWRITE)
      IF (IFLAG.LT.0.OR.IWRITE.NE.1) RETURN
C
C        THE DATA FILE DESCRIBING THE REDUCED SYSTEM IS PRINTED OUT
C
      CALL EQFORM (INB,NB,NLB,NEXT,NFIC,NBOUND,JBOUND,NRET,JRET,BTYP,BCV
     1,V,NEWNUM,IOLDNU,IBINTE,YT,JRYT,ICYT,LBINP,LBOUT,NTL,MODE3,INPT,IW
     2RITE)
      RETURN
   90 CONTINUE
      IFLAG=-5
      RETURN
  100 CONTINUE
      IFLAG=-6
      RETURN
  110 CONTINUE
      IFLAG=-7
      RETURN
  120 FORMAT (//,2X,"BOUNDARY BUSES: ",20(I4,1X))
  130 FORMAT (/,2X,"RETAINED BUSES:",20(I4,1X))
  140 FORMAT (2X,"THE SOLUTION FOR ",I4,1X,"BUS-SYSTEM"," USING THE STAN
     1DARD WARD METHOD WITHOUT SHUNTS")
  150 FORMAT (2X,"THE SOLUTION FOR ",I4,1X,"BUS-SYSTEM"," USING THE STAN
     1DARD WARD METHOD WITH BUFFER ZONE")
  160 FORMAT (2X,"THE SOLUTION FOR ",I4,1X,"BUS-SYSTEM"," USING THE EXTE
     1NDED WARD METHOD")
  170 FORMAT (2X,"THE SOLUTION FOR ",I4,1X,"BUS-SYSTEM"," USING THE SIMP
     1LIFIED EXTENDED WARD METHOD")
  180 FORMAT (2X,"THE SOLUTION FOR ",I4,1X,"BUS-SYSTEM"," USING THE STAN
     1DARD WARD METHOD WITH SHUNTS IN THE EXTERNAL SYSTEM")
  190 FORMAT (2X,"THE REDUCED POWER SYSTEM IS DETERMINED")
  200 FORMAT (2X,"THE EXTERNAL EQUIVALENT IS DETERMINED")
      END
C
C
      SUBROUTINE EQWEES (NB,NZ,NEXT,NBRE,NFIC,BTYP,JRYT,ICYT,YT,NROW,NCO
     1L,AK,VR,LCOL,ITAG,CE,NRB,AP,OTPT,IFLAG,IWRITE)
C
C        THIS SUBROUTINE CALCULATES THE NUMBER AND VALUES OF
C        FICTITIOUS BRANCHES
C
      INTEGER BTYP(1),JRYT(1),ICYT(1),NROW(1),NCOL(1),LCOL(1),ITAG(1),NR
     1B(1),OTPT
      COMPLEX Y,YT(1),AK(1),CE(1),AP(1)
C
      L=NEXT+NBRE
      KNEXT=NEXT
      J=0
```

```
C
C        THE LOCAL RENUMBERING OF BUSES IN THE EXTERNAL POWER SYSTEM
C        WITH GENERATOR BUSES GROUNDED. THE NEW INDICES OF THE BUSES ARE
C        STORED IN THE VECTOR NRB
C
      DO 40 I=1,L
      IF (I.GT.L) GO TO 30
      IF (BTYP(I).NE.0) GO TO 20
      J1=JRYT(I)+1
      J2=JRYT(I+1)-1
      DO 10 J3=J1,J2
      KK=ICYT(J3)
      IF (KK.GE.L) GO TO 10
      IF (BTYP(KK).EQ.0) GO TO 30
   10 CONTINUE
   20 CONTINUE
      NRB(I)=0
      IF (I.LE.NEXT) KNEXT=KNEXT-1
      GO TO 40
   30 CONTINUE
      J=J+1
      NRB(I)=J
   40 CONTINUE
      KB=J
C
C        END OF THE RENUMBERING
C
C        CALCULATE A COEFFICIENT MATRIX (VECTORS AK,NROW,NCOL) OF
C        THE EXTERNAL POWER SYSTEM WITH THE GENERATOR BUSES GROUNDED
C
      NZ=0
      DO 60 I=1,NEXT
C
C        IN THIS LOOP "NEXT" ROWS OF THE ADMITTANCE MATRIX ARE CHANGED
C        TO THE FORM OF THE COEFFICIENT MATRIX
C
      IF (NRB(I).EQ.0) GO TO 60
      J1=JRYT(I)
      J2=JRYT(I+1)-1
      DO 50 J=J1,J2
      KK=ICYT(J)
      IF (NRB(KK).EQ.0) GO TO 50
      NZ=NZ+1
      NROW(NZ)=NRB(I)
      NCOL(NZ)=NRB(KK)
      AK(NZ)=YT(J)
   50 CONTINUE
   60 CONTINUE
C
C        END OF THE CHANGES
C
      N=NEXT+1
      DO 80 I=N,L
C
C        IN THIS LOOP THE ROWS "N,...,L" OF THE ADMITTANCE MATRIX ARE
C        CHANGED TO THE FORM OF THE COEFFICIENT MATRIX
C
C        THE VALUES OF DIAGONAL ELEMENTS OF THE ADMITTANCE MATRIX OF THE
C        EXTERNAL SYSTEM CORRESPONDING TO THE BOUNDARY AND RETAINED BUSES
C        ARE CALCULATED
C
      IF (NRB(I).EQ.0) GO TO 80
      J1=JRYT(I)+1
      J2=JRYT(I+1)-1
      J3=JRYT(I)
```

```
      Y=(0.0,0.0)
      DO 70 J=J1,J2
      KK=ICYT(J)
      IF (NRB(KK).EQ.0) GO TO 70
      IF (KK.GE.L) GO TO 70
      NZ=NZ+1
      NROW(NZ)=NRB(I)
      NCOL(NZ)=NRB(KK)
      Y=Y+YT(J)
      AK(NZ)=YT(J)
   70 CONTINUE
C
C        THE DIAGONAL ELEMENTS
C
      NZ=NZ+1
      NROW(NZ)=NRB(I)
      NCOL(NZ)=NRB(I)
      AK(NZ)=-Y
   80 CONTINUE
C
C        END OF THE CHANGES
C
      DO 90 I=N,L
      IF (NRB(I).EQ.0.AND.BTYP(I).EQ.0) NRB(I)=-1
   90 CONTINUE
C
C        PERFORM THE GAUSSIAN ELIMINATION
C
      CALL GAUSEL (KB,NZ,KNEXT,NROW,NCOL,AK,VR,LCOL,ITAG,CE,IFLAG,IWRITE
     1)
      IF (IFLAG.LT.0) RETURN
C
C        CALCULATE A SPARSE ADMITTANCE MATRIX (VECTORS CE,LCOL,ITAG) OF
C        THE EXTERNAL EQUIVALENT
C
      KB=KB-KNEXT
      CALL SPAMAT (KB,NZ,NCOL,NROW,AK,LCOL,ITAG,CE)
C
C        CALCULATE THE NUMBER AND VALUES OF THE FICTITIOUS BRANCHES
C
      NFIC=0
      DO 130 I=1,NBRE
      J=NEXT+I
      J1=NRB(J)
      IF (J1.EQ.0) GO TO 130
      NFIC=NFIC+1
      Y=(0.0,0.0)
      IF (J1.LT.0) GO TO 110
      J1=LCOL(NFIC)+1
      J2=LCOL(NFIC+1)-1
      DO 100 J=J1,J2
      Y=Y+CE(J)
  100 CONTINUE
      J=LCOL(NFIC)
      Y=Y+CE(J)
      NRB(NFIC)=I
      AP(NFIC)=Y
      GO TO 130
  110 CONTINUE
      J1=JRYT(J)
      J2=J1+1
      J3=JRYT(J+1)-1
      DO 120 J=J2,J3
      IF (ICYT(J).GE.L) Y=Y+YT(J)
  120 CONTINUE
```

```
      NRB(NFIC)=I
      AP(NFIC)=YT(J1)+Y
  130 CONTINUE
C
C        THE VALUES OF THE FICTITIOUS BRANCHES ARE STORED IN THE VECTOR
C        AP. THE INDICES OF THE BOUNDARY BUSES INCIDENT WITH THE
C        FICTITIOUS BRANCHES ARE STORED IN THE VECTOR NRB
C
      IF (NFIC.EQ.0) RETURN
      IF (IWRITE.LT.3) RETURN
      WRITE (OTPT,150)
      DO 140 I=1,NFIC
      WRITE (OTPT,160) I,AP(I)
  140 CONTINUE
  150 FORMAT (//,4X,"THE FICTITIOUS BRANCH ADMITTANCES",//,4X,"BUS",13X,
     1"ADMITTANCE"/)
  160 FORMAT (2X,I4,4X,2(E12.5,2X))
      RETURN
      END
C
C
      SUBROUTINE GAUSEL (N,NEL,NEXT,NROW,NCOL,AK,V,LCOL,ITAG,CE,IFLAG,IW
     1RITE)
C
C        THIS SUBROUTINE PERFORMS GAUSSIAN ELIMINATION FOR A SYSTEM OF
C        LINEAR EQUATIONS DESCRIBED BY A COEFFICIENT MATRIX AK AND A
C        RIGHT HAND SIDE VECTOR V. THE ELIMINATION IS PERFORMED FOR A
C        NUMBER OF EQUATIONS GIVEN BY THE VARIABLE "NEXT"
C
C        N            NUMBER OF EQUATIONS
C        NEL          NUMBER OF NONZERO ELEMENTS IN THE COEFFICIENT MATRIX
C        NEXT         NUMBER OF EQUATIONS FOR WHICH THE ELIMINATION IS
C                     PERFORMED
C        NROW         VECTOR OF DIMENSION NEL. IT CONTAINS THE ROW INDICES
C                     OF ELEMENTS OF THE COEFFICIENT MATRIX
C        NCOL         VECTOR OF DIMENSION NEL. IT CONTAINS THE COLUMN
C                     INDICES OF ELEMENTS OF THE COEFFICIENT MATRIX
C        AK           VECTOR OF DIMENSION NEL. IT CONTAINS THE NONZERO
C                     ELEMENTS OF THE COEFFICIENT MATRIX
C        V            VECTOR OF DIMENSION NEL. RIGHT HAND SIDE VECTOR
C        LCOL         AUXILIARY VECTOR OF DIMENSION 2*N
C        ITAG         AUXILIARY VECTOR OF DIMENSION AT LEAST 2*NEL
C        CE           AUXILIARY VECTOR OF DIMENSION AT LEAST 2*NEL
C        IWRITE       PARAMETER THAT CONTROLS OUTPUT. IF IWRITE=4 THE
C                     COEFFICIENT MATRIX AK BEFORE AND AFTER GAUSSIAN
C                     ELIMINATION IS PRINTED OUT ON UNIT 6
C        IFLAG        RETURN FLAG FROM GAUSEL. IFLAG=0 FOR A SUCCEFUL
C                     RETURN. IFLAG=-1 IF THE LENGTH OF VECTORS ITAG AND CE
C                     HAVE BEEN DECLARED TOO SMALL
C
      INTEGER NROW(1),NCOL(1),LCOL(1),ITAG(1),OTPT
      COMPLEX CE(1),AK(1),V(1)
      OTPT=6
C
      CALL SPAMAT (N,NEL,NCOL,NROW,AK,LCOL,ITAG,CE)
C
      CALL ASUB (N,NEXT,NEL,LCOL,ITAG,CE,V,AK,NCOL,NROW,IFLAG,IWRITE)
C
      RETURN
      END
C
C
      SUBROUTINE INTERV (ISTOS,IBOUND,IBINTE,NINTE,JRGA,ICGA)
C
C        THIS SUBROUTINE DETERMINES THE BUSES BELONGING TO THE INTERNAL
```

```
C        POWER SYSTEM. THE RESULTS ARE STORED IN THE VECTOR IBINTE.
C        IBINTE(K) IS SET TO 1 IF THE KTH BUS BELONGS TO THE INTERNAL
C        POWER SYSTEM. THE NUMBER OF THE INTERNAL BUSES IS STORED AS THE
C        PARAMETER NINTE
C
C        ISTOS       AN AUXILIARY VECTOR
C        JRGA        VECTOR CONTAINS ROW INDICES OF THE ADJACENCY
C                    MATRIX
C        ICGA        VECTOR CONTAINS COLUMN INDICES OF THE ADJACENCY
C                    MATRIX
C        IBOUND      VECTOR CONTAINS THE CODED INDICES OF THE BUSES
C                    BELONGING TO THE BOUNDARY
C
      INTEGER ISTOS(1),IBOUND(1),IBINTE(1),JRGA(1),ICGA(1)
      LOGICAL BELONG
C
      ICSTOS=1
      NINTE=1
C
C        ISTOS(1) CONTAINS THE INDEX OF THE SLACK BUS
C
      IBINTE(ISTOS(1))=1
   10 CONTINUE
      N=ISTOS(ICSTOS)
C
C        DETERMINE THE INDICES OF BUSES INCIDENT WITH THE NTH BUS
C
      ICSTOS=ICSTOS-1
      INC1=JRGA(N)+1
      INC2=JRGA(N+1)
      DO 20 J=INC1,INC2
      INCJ=ICGA(J)
      IF (BELONG(INCJ,IBOUND)) GO TO 20
      IF (IBINTE(INCJ).EQ.1) GO TO 20
C
C        THE INCJTH BUS BELONGS TO THE INTERNAL SYSTEM
C
      IBINTE(INCJ)=1
      NINTE=NINTE+1
C
C        PUT THE INDEX INCJ ONTO THE STACK (VECTOR ISTOS)
C
      ICSTOS=ICSTOS+1
      ISTOS(ICSTOS)=INCJ
   20 CONTINUE
      IF (ICSTOS.GT.0) GO TO 10
      RETURN
      END
C
C
      SUBROUTINE MOVE (N,NYT,YT,JRYT,ICYT)
C
C        THIS SUBROUTINE REMOVES THE FIRST ROW FROM A MATRIX STORED IN
C        SPARSE FORM (VECTORS YT,JRYT,ICYT) AND MODIFIES THESE VECTORS
C
      INTEGER JRYT(1),ICYT(1)
      COMPLEX YT(1)
C
      K=2
C
C        MODIFY THE VECTORS ICYT AND YT
C
      NEL=JRYT(K)-1
      NYT=NYT-NEL
      DO 10 I=1,NYT
```

```
      K=NEL+I
      ICYT(I)=ICYT(K)
      YT(I)=YT(K)
   10 CONTINUE
C
C        MODIFY THE VECTOR JRYT
C
      DO 20 I=2,N
      JRYT(I)=JRYT(I+1)-NEL
   20 CONTINUE
C
C        MODIFY THE VECTOR ICYT
C
      DO 30 I=1,NYT
      ICYT(I)=ICYT(I)-1
   30 CONTINUE
      N=N-1
      RETURN
      END
C
C
      SUBROUTINE NOTE (IEL,IB)
C
C        THIS SUBROUTINE CODES THE ELEMENT IEL IN THE VECTOR IB
C
      INTEGER IB(1)
C
      IX=(IEL-1)/50
      IY=IEL-IX*50
      J=IX+1
      CALL SETBIT (IB(J),IY)
      RETURN
      END
C
C
      SUBROUTINE ORDEVE (NB,NTL,LBINP,LBOUT,JRGA,ICGA)
C
C        THIS SUBROUTINE FORMS AN ADJACENCY MATRIX OF A GRAPH AND STORES
C        IT IN A SPARSE FORM (VECTORS JRGA AND ICGA)
C        NB          NUMBER OF NODES
C        NTL         NUMBER OF EDGES
C        LBINP,LBOUT  INTEGER VECTORS OF DIMENSION NTL. LBINP(K),LBOUT(K)
C                     STORE THE NODES OF THE GRAPH INCIDENT WITH THE KTH
C                     EDGE
C        JRGA        STARTING POSITIONS OF ROWS. JRGA(K)+1 INDICATES THE
C                    FIRST POSITION OF ELEMENTS OF THE KTH ROW IN THE
C                    VECTOR ICGA
C        ICGA        STORES THE COLUMN INDICES OF NONZERO ELEMENTS OF THE
C                    ADJACENCY MATRIX
C
      INTEGER LBINP(1),LBOUT(1),JRGA(1),ICGA(1)
C
      J=NB+1
C
      DO 10 I=1,J
      JRGA(I)=0
   10 CONTINUE
C
C        CALCULATE THE NUMBER OF EDGES INCIDENT WITH EACH NODE
C
      DO 20 I=1,NTL
      JRGA(LBINP(I))=JRGA(LBINP(I))+1
      JRGA(LBOUT(I))=JRGA(LBOUT(I))+1
   20 CONTINUE
C
```

```
C        CALCULATE THE VECTOR JRGA
C
      NS2=JRGA(1)
      JRGA(1)=1
      DO 30 I=1,NB
      NS1=JRGA(I+1)
      JRGA(I+1)=JRGA(I)+NS2
      NS2=NS1
   30 CONTINUE
C
C        CALCULATE THE VECTOR ICGA
C
      DO 40 I=1,NTL
      NS1=LBINP(I)
      NS2=LBOUT(I)
      ICGA(JRGA(NS1))=NS2
      ICGA(JRGA(NS2))=NS1
      JRGA(NS1)=JRGA(NS1)+1
      JRGA(NS2)=JRGA(NS2)+1
   40 CONTINUE
C
C        MODIFICATION OF THE VECTOR JRGA
C
      NS1=JRGA(1)
      JRGA(1)=0
      IX=NB+1
      DO 50 I=2,IX
      NS2=JRGA(I)
      JRGA(I)=NS1-1
      NS1=NS2
   50 CONTINUE
      RETURN
      END
C
C
      SUBROUTINE PRINAD (N,JRYT,ICYT,YT,IWRITE,OTPT)
C
C        THIS  SUBROUTINE PRINTS OUT AN ADMITTANCE MATRIX STORED IN A
C        SPARSE FORM (VECTORS YT,JRYT,ICYT)
C
      INTEGER JRYT(1),ICYT(1),OTPT
      COMPLEX YT(1)
C
      IF (IWRITE.NE.3) RETURN
      WRITE (OTPT,40)
      DO 10 I=1,N
      J1=JRYT(I)
      J2=JRYT(I+1)-1
      WRITE (OTPT,20) ICYT(J1)
      WRITE (OTPT,30) (ICYT(J),YT(J),J=J1,J2)
   10 CONTINUE
   20 FORMAT (//"BUS NO.",I3)
   30 FORMAT (3(3X,I3,":",2(1X,E13.7)))
   40 FORMAT (//" ADMITTANCE MATRIX YT"//"IN EACH ROW OF MATRIX ","DATA
     1IS IN SEQUENCE: COLUMN NUMBER,REAL(YT),IMAG(YT)")
      RETURN
      END
C
C
      SUBROUTINE PRINTNC (N,NROW,NCOL,V,OTPT)
C
C        THIS SUBROUTINE PRINTS THE NONZERO COEFFICIENTS OF A COMPLEX
C        MATRIX V
C
      INTEGER NROW(1),NCOL(1),OTPT
```

```
      COMPLEX V(1)
C
      IF (N.LE.0) RETURN
      IF (N.GT.1) GO TO 10
      WRITE (OTPT,110)
      WRITE (OTPT,100)
      GO TO 30
   10 CONTINUE
      IF (N.EQ.4.OR.N.EQ.2) GO TO 20
      WRITE (OTPT,80)
      WRITE (OTPT,50)
      GO TO 30
   20 WRITE (OTPT,70)
      WRITE (OTPT,60)
   30 N3=(N+2)/3
      DO 40 I=1,N3
      LI=I
      LO=I+2*N3
      IF (LO.GT.N) LO=I+N3
      IF (LO.GT.N) LO=I
      WRITE (OTPT,90) (NROW(J),NCOL(J),V(J),J=LI,LO,N3)
   40 CONTINUE
      RETURN
   50 FORMAT (1H ,11X,"       REAL            IMAGINARY",18X,"REAL           IM
     1AGINARY",18X,"REAL           IMAGINARY",/)
   60 FORMAT (1H ,11X,"       REAL            IMAGINARY",18X,"REAL           IM
     1AGINARY",/)
   70 FORMAT (1H ," ROW  COL  ",11X,"VALUE                         ROW  COL",11X,"
     1VALUE",/)
   80 FORMAT (1H ," ROW  COL  ",11X,"VALUE                         ROW  COL  ",11X
     1,"VALUE                 ROW  COL  ",11X,"VALUE",/)
   90 FORMAT (1X,I4,I5,2E15.6,2I5,2E15.6,2I5,2E15.6)
  100 FORMAT (1H ,16X,"REAL",9X,"IMAGINARY",/)
  110 FORMAT (1H ," ROW  COL ",14X,"VALUE",/)
      END
C
C
      SUBROUTINE RDATA (NBOUND,JBOUND,NRET,JRET,INPT)
C
C         THIS SUBROUTINE READS THE DATA DESCRIBING BOUNDARY AND RETAINED
C         BUSES
C
      INTEGER JBOUND(1),JRET(1)
C
C         READ THE NUMBER OF BOUNDARY BUSES
C
      READ (INPT,*) NBOUND
C
C         READ BOUNDARY BUSES
C
      DO 10 I=1,NBOUND
      READ (INPT,*) JBOUND(I)
   10 CONTINUE
C
C         READ THE NUMBER OF RETAINED BUSES
C
      READ (INPT,*) NRET
      IF (NRET.EQ.0) RETURN
C
C         READ RETAINED BUSES
C
      DO 20 I=1,NRET
      READ (INPT,*) JRET(I)
   20 CONTINUE
C
```

```
      RETURN
      END
C
C
      SUBROUTINE SETBIN (NBI,JBI,IBI)
C
C       THIS SUBROUTINE CODES NBI ELEMENTS OF THE VECTOR JBI IN
C       THE VECTOR IBI
C
      INTEGER JBI(1),IBI(1)
C
      DO 10 I=1,NBI
      CALL NOTE (JBI(I),IBI)
   10 CONTINUE
      RETURN
      END
C
C
      SUBROUTINE SETBIT (K,I)
C
C       THIS SUBROUTINE SETS TO 1 THE  ITH BIT IN THE WORD K
C
      J=1
      II=I-1
      J=SHIFT(J,II)
      K=K.OR.J
      RETURN
      END
C
C
      SUBROUTINE SETVEC (N,NYT,JRYT,ICYT,YT,I,J,NEL,IFLAG)
C
C       IN THIS SUBROUTINE THE ITH ROW OF THE MATRIX STORED IN THE
C       SPARSE FORM (VECTORS YT,JRYT,ICYT) IS SUBTRACTED FROM THE JTH
C       ROW OF THIS MATRIX
C
      INTEGER ICYT(1),JRYT(1)
      COMPLEX YT(1)
C
C       SUBTRACT THE ITH ROW FROM THE JTH ROW
C
      I1=JRYT(I)+1
      I2=JRYT(I+1)-1
      J1=JRYT(J)
      J2=JRYT(J+1)-1
      NEL=0
      IFLAG=0
      NEL1=NYT
      DO 10 J3=J1,J2
      IF (I.EQ.ICYT(J3)) GO TO 20
   10 CONTINUE
      IFLAG=-10
      WRITE (6,160) IFLAG
      RETURN
   20 CONTINUE
C
C       CALCULATE THE NUMBER OF THE NEW ELEMENTS IN THE JTH ROW
C
      INEL=J3
      DO 40 I3=I1,I2
      I4=ICYT(I3)
      DO 30 J3=J1,J2
      IF (I4.EQ.ICYT(J3)) GO TO 40
   30 CONTINUE
      NEL=NEL+1
```

```
   40 CONTINUE
C
C         MODIFY MATRICES YT,JRYT,ICYT (CHANGE THE LENGTH)
C
      JNEL=JRYT(J+1)-1
      IF (NEL.EQ.0) GO TO 70
      IF (NEL.NE.0) NEL=NEL-1
      IF (NEL.EQ.0) GO TO 100
      I1=NYT+1
      I2=NEL+I1
      J1=0
      J2=JNEL+1
      DO 50 J3=J2,NYT
      J1=J1+1
      ICYT(I2-J1)=ICYT(I1-J1)
      YT(I2-J1)=YT(I1-J1)
   50 CONTINUE
      I2=N+1
      I1=J+1
      DO 60 I3=I1,I2
      JRYT(I3)=JRYT(I3)+NEL
   60 CONTINUE
      NYT=NYT+NEL
      GO TO 100
   70 CONTINUE
      NYT=NYT-1
      DO 80 I1=INEL,NYT
      ICYT(I1)=ICYT(I1+1)
      YT(I1)=YT(I1+1)
   80 CONTINUE
      I1=J+1
      I2=N+1
      DO 90 I3=I1,I2
      JRYT(I3)=JRYT(I3)-1
   90 CONTINUE
      JNEL=JNEL-1
  100 CONTINUE
      IF (NYT.LE.2*NEL1) GO TO 110
      IFLAG=-1
      RETURN
  110 CONTINUE
C
C       CALCULATE THE NEW VALUES OF THE ELEMENTS OF THE JTH ROW
C
      K=0
      I1=JRYT(I)+1
      I2=JRYT(I+1)-1
      J1=JRYT(J)
      J2=JRYT(J+1)-1
C
      DO 150 I3=I1,I2
      I4=ICYT(I3)
      DO 120 J3=J1,JNEL
      IF (I4.EQ.ICYT(J3)) GO TO 140
  120 CONTINUE
      IF (K.EQ.NEL) GO TO 130
      ICYT(J2-K)=I4
      YT(J2-K)=-YT(I3)
      K=K+1
      GO TO 150
  130 CONTINUE
      ICYT(INEL)=I4
      YT(INEL)=-YT(I3)
      GO TO 150
  140 CONTINUE
```

```
      YT(J3)=YT(J3)-YT(I3)
  150 CONTINUE
      RETURN
  160 FORMAT (2X,"IFLAG FROM SETVEC IFLAG= ",I3,3X,"AN ERROR IN THE MATR
     1IX ICYT")
      END
C
C
      SUBROUTINE SPAMAT (N,NEL,NCOL,NROW,AK,JRYT,ICYT,YT)
C
C        THIS SUBROUTINE CHANGES THE COEFFICIENT MATRIX (VECTORS AK,NROW,
C        NCOL) TO THE SPARSE MATRIX FORM AND STORES IT IN VECTORS YT,JRYT
C        AND ICYT
C
      COMPLEX AK(1),YT(1),Y,Y1
      INTEGER NCOL(1),NROW(1),JRYT(1),ICYT(1)
C
      N1=N+1
      DO 10 I=1,N1
      JRYT(I)=0
   10 CONTINUE
      JRYT(1)=1
C
C        CALCULATE THE NUMBER OF ELEMENTS IN EACH ROW
C
      DO 20 I=1,NEL
      JRYT(NROW(I)+1)=JRYT(NROW(I)+1)+1
   20 CONTINUE
C
C        MODIFY THE VECTOR JRYT
C
      DO 30 I=1,N
      JRYT(I+1)=JRYT(I)+JRYT(I+1)
   30 CONTINUE
C
C        CALCULATE THE VECTORS YT,ICYT
C
      DO 40 I=1,NEL
      NRWI=NROW(I)
      NCLI=NCOL(I)
      LCNC=JRYT(NRWI)
      YT(LCNC)=AK(I)
      ICYT(LCNC)=NCLI
      JRYT(NRWI)=LCNC+1
   40 CONTINUE
C
C        MODIFY THE VECTORS JRYT,ICYT
C
      DO 50 I=1,N
      J=N-I+2
      JRYT(J)=JRYT(J-1)
   50 CONTINUE
      JRYT(1)=1
      DO 60 I=1,N
      IF (JRYT(I).EQ.JRYT(I+1)) JRYT(I)=0
   60 CONTINUE
      DO 90 I=1,N
C
C        CHANGE THE POSITION OF DIAGONAL ELEMENTS IN THE VECTOR ICYT
C
      J1=JRYT(I)
      J2=JRYT(I+1)-1
      DO 80 J=J1,J2
      J3=ICYT(J)
      IF (J3.EQ.I) GO TO 70
```

```
      GO TO 80
   70 CONTINUE
      Y=YT(J)
      Y1=YT(J1)
      YT(J)=Y1
      YT(J1)=Y
      ICYT(J)=ICYT(J1)
      ICYT(J1)=J3
      GO TO 90
   80 CONTINUE
   90 CONTINUE
      RETURN
      END
C
C
      SUBROUTINE ZERO (N,IB)
C
C       THIS SUBROUTINE SETS TO ZERO N ELEMENTS OF THE VECTOR IB
C
      INTEGER IB(1)
C
      DO 10 I=1,N
      IB(I)=0
   10 CONTINUE
      RETURN
      END
```